



Agenzia Nazionale per le Nuove Tecnologie,  
l'Energia e lo Sviluppo Economico Sostenibile



*Ministero dello Sviluppo Economico*

RICERCA DI SISTEMA ELETTRICO

Sviluppo del codice strutturale della catena francese CAST3M

*G.M. Giannuzzi*

SVILUPPO DEL CODICE STRUTTURALE DELLA CATENA FRANCESE CAST3M

G.M. Giannuzzi (ENEA)

Settembre 2011

Report Ricerca di Sistema Elettrico

Accordo di Programma Ministero dello Sviluppo Economico – ENEA

Area: Governo, Gestione e sviluppo del sistema elettrico nazionale

Progetto: Nuovo nucleare da fissione: collaborazioni internazionali e sviluppo competenze in materia nucleare

Responsabile Progetto: Paride Meloni, ENEA



**Titolo**

**Sviluppo del codice strutturale della catena francese CAST3M**

(Aggiornamento della procedura HTCTTRAN per il trasferimento di massa e di calore nei calcestruzzi)

**Descrittori**

Tipologia del documento: Rapporto Tecnico

Collocazione contrattuale: Accordo di programma ENEA-MSE: tema di ricerca "Nuovo nucleare da fissione"

Argomenti trattati: Trasmissione del calore e di massa nel calcestruzzo.  
Software agli elementi finiti.

**Sommario**

Nel presente rapporto si evidenziano le attività svolte nel secondo periodo di consuntivazione nell'ambito dell'azione CAST3M, che prevede l'aggiornamento della procedura HTCTTRAN per la simulazione dei fenomeni di trasporto di massa e di calore nei calcestruzzi, basata sul modello proposto da Bazant. Nella riorganizzazione e nel potenziamento del software, in accordo con i responsabili lato CEA del sistema CAST3M, si è scelta la filosofia di conservare per la procedura HTCTTRAN la caratteristica di strumento ingegneristico, in grado di trattare con buona approssimazione, gli scenari presenti negli edifici di contenimento di reattori nucleari (con esclusione dell'interazione corium-calcestruzzo), nei depositi di stoccaggio di rifiuti radioattivi a forte attività e nei sistemi di accumulo del calore per impianti solari termodinamici, casi in cui si hanno temperature non eccedenti significativamente i 600 °C.

**Note**

Copia n.

In carico a:

2			NOME			
			FIRMA			
1			NOME			
			FIRMA			
0	EMISSIONE	15/4/2011	NOME	G.M. Giannuzzi	M. Sepielli	P. MELONI
			FIRMA			
REV.	DESCRIZIONE	DATA	REDAZIONE	CONVALIDA	APPROVAZIONE	

 <b>Ricerca Sistema Elettrico</b>	<b>Sigla di identificazione</b>	<b>Rev.</b>	<b>Distrib.</b>	<b>Pag.</b>	<b>di</b>
	NNFISS – LP1 - 019	0	L	2	59

## SOMMARIO

<b>1</b>	<b>INTRODUZIONE .....</b>	<b>3</b>
<b>2</b>	<b>IL MODELLO DI BAZANT-THONGUTHAI E SUA IMPLEMENTAZIONE NUMERICA IN HTCTRAN .....</b>	<b>4</b>
2.1	EQUAZIONI PER I BILANCI DI MASSA E DI CALORE .....	4
2.2	CARATTERISTICHE DELL'IMPLEMENTAZIONE NUMERICA IN LINGUAGGIO GIBIANE .....	5
<b>3</b>	<b>DESCRIZIONE DEGLI AGGIORNAMENTI .....</b>	<b>6</b>
3.1	CORRELAZIONE PER LA DEIDRATAZIONE DELL'ACQUA LEGATA .....	6
3.2	ESTENSIONE CAPACITÀ APPARENTE DEL CALCESTRUZZO .....	6
3.3	VARIAZIONE DELLA CONDUCEBILITÀ TERMICA CON LA TEMPERATURE E IL CONTENUTO D'ACQUA LIBERA .....	7
3.4	VELOCIZZAZIONE DELLA PROCEDURA HTC_PER E HTC_WWW .....	8
3.5	ESTENSIONE A DOMINI TRIDIMENSIONALI .....	9
3.6	MIGLIORAMENTI FUNZIONALI .....	9
<b>4</b>	<b>TEST DI CONVALIDA .....</b>	<b>9</b>
4.1	MODULO DI ACCUMULO .....	10
4.2	PROVINO CUBICO TRIDIMENSIONALE .....	14
<b>5</b>	<b>CONCLUSIONI .....</b>	<b>17</b>
<b>6</b>	<b>ALLEGATO 1 :LISTATO DELLA PROCEDURA HTCTRAN E PROCEDURE AD ESSA COLLEGATE .....</b>	<b>19</b>
<b>7</b>	<b>ALLEGATO 2: FILE DI INPUT CASI DI CONVALIDA .....</b>	<b>49</b>
7.1	CASO-W1 MODULO DI ACCUMULO .....	49
7.2	CASO-K2 PROVINO CUBICO TRIDIMENSIONALE .....	53
<b>8</b>	<b>BIBLIOGRAFIA .....</b>	<b>58</b>

 <b>Ricerca Sistema Elettrico</b>	<b>Sigla di identificazione</b>	<b>Rev.</b>	<b>Distrib.</b>	<b>Pag.</b>	<b>di</b>
	NNFISS – LP1 - 019	0	L	3	59


## **1 INTRODUZIONE**

Le attività svolte nell'ambito dell'azione CAST3M prevedono l'aggiornamento della procedura HTCTTRAN per la simulazione dei fenomeni di trasporto di massa e di calore nei calcestruzzi, basata sul modello proposto da Bazant. La procedura HTCTTRAN, sviluppata in ENEA nel 1998, ha subito negli anni solo alcune modifiche per renderla compatibile con mutamenti apportati agli <operatori> presenti nel linguaggio <Gibiane> di CAST3M; ma la struttura di base e le limitazioni, presenti nello sviluppo originario, sono rimaste immutate.

La necessità di affrontare studi ed analisi di complessità maggiore sia in ENEA che al CEA, ha richiesto una riorganizzazione e potenziamento del software. In accordo con i responsabili, lato CEA, del sistema CAST3M, si è scelta la filosofia di conservare per la procedura HTCTTRAN la caratteristica di strumento ingegneristico, in grado di trattare con buona approssimazione, gli scenari presenti negli edifici di contenimento di reattori nucleari (con esclusione dell'interazione corium-calcestruzzo), nei depositi di stoccaggio di rifiuti radioattivi a forte attività e nei sistemi di accumulo del calore per impianti solari termodinamici, casi in cui si hanno temperature non eccedenti significativamente i 600 °C. Il superamento di tali temperature nel calcestruzzo, coinvolge fenomeni fisici, come la decarbonatazione di aggregati calcarei o la fusione dei costituenti presenti nel calcestruzzo non compatibili con il modello proposto da Bazant e comunque non trattabili in maniera unificata su tutto l'intervallo di temperature.

In particolare si è evidenziata la necessità di effettuare le seguenti estensioni del funzionamento:

- Introdurre la possibilità di trattare domini tridimensionali, in aggiunta al caso piano ed assialsimmetrico.
- Introdurre la variazione della conducibilità termica, con il contenuto di umidità e con la temperatura, tenendo in conto l'effetto memoria come evidenziato dalle misure sperimentali disponibili in letteratura.
- Aggiornare le correlazioni per la stima del rilascio dell'acqua legata (deidratazione), introducendo la possibilità di tener conto del grado di idratazione del calcestruzzo, del contenuto di cemento e dell'irreversibilità della deidratazione al ricalcare della temperatura.
- Tener conto del cambiamento della capacità termica apparente del calcestruzzo al variare della temperatura a seguito della deidratazione.
- Velocizzare gli algoritmi risolutivi, ridurre la taglia della memoria richiesta, migliorare la leggibilità del sorgente, consentire una maggior controllo della convergenza sempre a rischio in presenza di forti nonlinearità presenti nei parametri del modello.

 <b>Ricerca Sistema Elettrico</b>	<b>Sigla di identificazione</b>	<b>Rev.</b>	<b>Distrib.</b>	<b>Pag.</b>	<b>di</b>
	NNFISS – LP1 - 019	0	L	4	59

## 2 IL MODELLO DI BAZANT-THONGUTHAI E SUA IMPLEMENTAZIONE NUMERICA IN HTCTRAN

Il modello proposto da BAZANT-THONGUTHAI per il trasferimento di massa e calore nei calcestruzzi, implementato nella procedura HTCTRAN, si basa su un approccio fenomenologico che cerca di limitare opportunamente, senza perdita di capacità predittiva, l'estesa mole di dati sulle caratteristiche fisiche e meccaniche, richieste da altri modelli che entrano più in profondità nel comportamento fisico del calcestruzzo.

Le caratteristiche principali del modello di Bazant sono le seguenti:

- l'acqua libera in fase liquida e vapore è trattata con una sola fase omogeneizzata attraverso la variabile <free water content> (contenuto di acqua evaporabile per unità di volume con esclusione dell'acqua legata chimicamente alla fase solida),
- trasporto di calore per diffusione e convezione,
- trasporto di massa (acqua libera) secondo la legge di Darcy,
- formulazione semi-empirica per la permeabilità e l'assorbimento dell'acqua,
- deidratazione istantanea dell'acqua legata
- equilibrio termico tra la fase fluida e dello scheletro solido
- discretizzazione delle equazioni di bilancio secondo il metodo degli elementi finiti nello spazio, mentre nel tempo si è adottata una discretizzazione a differenze finite secondo il  $\alpha$ -method.

Rispetto al modello standard di Bazant, si è introdotto in via semplificata l'effetto della vaporizzazione interna al calcestruzzo e l'arresto della deidratazione in presenza di saturazione.

In quanto segue richiameremo le equazioni che governano il trasferimento di massa e di calore e le correlazioni per la determinazione della permeabilità relativa e delle isoterme di assorbimento che regolano lo stoccaggio dell'acqua libera nel calcestruzzo. Per una trattazione più estesa si rimanda ai documenti riportati in bibliografia [1] e [2] relativi alla precedente implementazione, e [3], [4] e [5] per quanto riguarda il modello di Bazant.

### 2.1 EQUAZIONI PER I BILANCI DI MASSA E DI CALORE

L'equazione impiegata per il bilancio di massa della fase fluida composta da un mix vapore ed acqua allo stato liquido.

$$\frac{\partial W}{\partial t} - \frac{\partial W_d}{\partial t} = -\text{div}J$$

in cui il termine diffusivo è derivato in base alla legge di Darcy

$$J = -\frac{a}{g} \text{grad}P$$

W rappresenta il contenuto d'acqua libera,  $W_d$  l'acqua legata rilasciata nel processo di deidratazione indotto dal riscaldamento e la sua derivata determina il conseguente termine di sorgente che va ad incrementare il contenuto di acqua libera,  $a$  e  $g$  rispettivamente la permeabilità relativa e l'accelerazione di gravità.

L'altra equazione da abbinare alla precedente riguarda il bilancio di calore

$$\rho C \frac{\partial T}{\partial t} - C_a \frac{\partial w}{\partial t} - C_w J \cdot \text{grad} T = -\text{div} q$$

$$\text{con } q = -k \text{ grad } T$$

in cui oltre ai classici termini, capacitivo e diffusivo, compaiono il termine legato alla vaporizzazione dell'acqua libera e il termine convettivo che rappresenta il calore trasportato dal movimento del fluido.  $C_a$  e  $C_w$  rappresentano nell'ordine il calore di vaporizzazione e il calore specifico dell'acqua libera.

Considerando che  $W(P,T)$  è funzione della pressione interna e della temperatura del fluido e che  $W_d$  è funzione della sola temperatura l'equazione precedente assume la forma

$$\frac{\partial w}{\partial P} \frac{\partial P}{\partial t} + \frac{\partial w}{\partial T} \frac{\partial T}{\partial t} - \frac{\partial w_d}{\partial T} \frac{\partial T}{\partial t} = -\text{div} \left( -\frac{a}{g} \text{grad} P \right)$$

e

$$\left( \rho_s C_s + C_d \frac{\partial w_d}{\partial T} + w C_w \right) \frac{\partial T}{\partial t} - C_a \left[ \frac{\partial w}{\partial T} \frac{\partial T}{\partial t} + \frac{\partial w}{\partial P} \frac{\partial P}{\partial t} \right] + C_w \frac{a}{g} \text{grad} P \cdot \text{grad} T = \text{div} (k \text{ grad} T)$$

Come si vede l'equazione di bilancio contiene dei termini misti in P e T che dovranno essere opportunamente separati per consentire l'implementazione della stessa in CAST3M con gli <operatori gibiane> esistenti che non prevedono la contemporanea presenza della variabile pressione e temperatura.

## 2.2 CARATTERISTICHE DELL'IMPLEMENTAZIONE NUMERICA IN LINGUAGGIO GIBIANE

Nel linguaggio gibiane sono presenti operatori differenziali e per la costruzione dei termini di sorgente tipici dell'equazione denominata quasi armonica che è alla base del trasporto di calore. Mancando i termini accoppiati pressione e temperatura, è stata impiegata una procedura cosiddetta <staggered> [10], un modo da ottenere attraverso un opportuno processo iterativo il necessario accoppiamento tra le equazioni di bilancio. I termini misti vengono spostati sul lato destro delle equazioni di bilancio e trattati analogamente agli usuali carichi termici e idrici, ritrasformando così le equazioni risolventi in equazioni quasi armoniche trattabili direttamente in CAST3M. La maggior delicatezza del processo iterativo di convergenza è però largamente il risparmio nello sviluppo diretto in Fortran degli operatori misti.

Le equazioni di bilancio, discretizzate nello spazio secondo il metodo degli elementi finiti e nel tempo secondo lo schema a differenze finite noto come  $q_1$ -method ([15] 85) danno origine al sistema risolvibile riportato di seguito:

$$\begin{cases} (Ct^* / \Delta t + Kt^* \theta + Ktb) T_{n+1}^i = [Ct^* / \Delta t - (1 - \theta) Kt^*] T_n + Ft^* \\ (Cp^* / \Delta t + Kp^* \theta + Kpb) P_{n+1}^i = [Cp^* / \Delta t - (1 - \theta) Kp^*] P_n + Fp^* \end{cases}$$

dove Ct e Cp rappresentano i termini capacitivi, Kt e Kp i termini conduttivi, Ktb e Kpb le rigidzze associate ai vincoli imposti, Ft e Fp i carichi; le grandezze asteriscate indicano che sono calcolate alla temperatura  $T^*$  e pressione  $P^*$ , valori intermedi tra i valori relativi al passo precedente e al passo attuale iterazione precedente, in funzione del parametro  $\alpha$  scelto. In valore di default assunto per  $\alpha$  è pari a 0.55 che dà origine allo schema Crank-Nicolson modificato.

### 3 DESCRIZIONE DEGLI AGGIORNAMENTI

#### 3.1 CORRELAZIONE PER LA DEIDRATAZIONE DELL'ACQUA LEGATA

Nella versione 1998 si prevedeva l'inserimento in forma tabellare della quantità di acqua rilasciata nel processo di deidratazione in funzione della temperatura, a cura dell'utilizzatore o si poteva utilizzare una tabella di default presa da [4] modulabile con il contenuto di cemento del mix. Attualmente si è inserita la seguente correlazione, già impiegata sperimentata nel codice HITECOSP2 ( [8] [9] ) :

$$W_d = \frac{1}{2} F_{ste} \cdot F_{idr} \cdot C \cdot \left[ 1 + \operatorname{sen} \left( \frac{\pi}{2} (1 - 2 \exp(-0.004(T - 105))) \right) \right]$$

Essa consente di modulare il rilascio di acqua legata con il fattore stechiometrico, che dà quanta acqua è necessaria per idratare l'unità di massa di cemento, il fattore di idratazione che indica la quantità di cemento idratato e con la quantità di cemento per unità di volume di calcestruzzo.

La deidratazione resta permanente ovviamente al ricalare della temperatura.

#### 3.2 ESTENSIONE CAPACITÀ APPARENTE DEL CALCESTRUZZO

Nella capacità termica <effettiva> del calcestruzzo vengono conglobati oltre i contributo relativi ai calori sensibili dei materiali in esso contenuti (acqua, sabbia, aggregato e cemento anidro) e del calore di deidratazione; si è aggiunto nella formula già impiegata un quarto termine per tener in conto della diminuita capacità termica del calcestruzzo secco (privo di acqua libera), conseguente alla deidratazione avvenuta e sostituita la densità del secco con  $m_s$  che rappresenta il contenuto degli aggregati e del solo cemento anidro.

$$(\rho C)_{eff} = m_s C_s + W C_w + C_d \frac{\partial W_d}{\partial T} + (W_{do} - W_d) \cdot C_{bw}$$

In questo modo la capacità termica del secco viene modulata in funzione del grado di deidratazione.  $C_{bw}$  rappresenta il calore specifico dell'acqua interstiziale e legata che secondo [6] risulta pari a 3.76 kJ/°C-Kg, leggermente inferiore a quello dell'acqua libera ( $C_w$ );  $W_{do}$  è il contenuto di acqua legata iniziale prima del riscaldamento. La variazione di  $C_s$  con la temperatura in genere non viene considerata perché piccola;



si ricorda che trattandosi di una miscela essa può essere valutata conoscendo i calori specifici e le frazioni di massa  $w_i$  di ciascun componente nel modo seguente:

$$C = \sum_i w_i C_i$$

### 3.3 VARIAZIONE DELLA CONDUCIBILITÀ TERMICA CON LA TEMPERATURE E IL CONTENUTO D'ACQUA LIBERA

La conducibilità nella precedente versione veniva considerata costante. Attualmente si è implementato un modello più realistico che prevede per la conducibilità del calcestruzzo secco

$$\lambda_d = \lambda_{d0} [1 + A_\lambda (T_{\max} - T_0)]$$

e che valuta la conducibilità effettiva tenendo del contenuto di acqua libera nel modo seguente

$$\lambda_{eff} = \lambda_d \cdot \left( 1 + 4 \frac{W}{m_s} \right)$$

Tale modello, già impiegato nel codice HITECOSP2 [8] è stato proposto da Schrefler per i calcestruzzi; ulteriori approfondimenti sull'identificazioni dei parametri di trasporto nei calcestruzzi possono essere ricavati nei riferimenti [11] [12] e [13].




Figura 1 Variazione della conduttività termica in funzione della temperatura (Marechal [14] ) e schematizzazione del comportamento al raffreddamento.

La variazione della conducibilità del secco in fase di raffreddamento viene mantenuta costante, con valore pari a quello raggiunto al massimo della fase riscaldamento come evidenziato in Figura 1.

La modellizzazione del calo di conducibilità durante il raffreddamento che si manifesta soprattutto a temperature di oltre 250 °C, probabilmente dovuta al distacco per ritiro differenziale dell'aggregato rispetto alla pasta, al momento risulta piuttosto complesso da realizzare in condizioni di ciclaggio termico; va detto che in quasi tutti i modelli implementati, per il comportamento in temperatura di calcestruzzi, non viene considerata l'irreversibilità con la temperatura, essendo quasi sempre strumenti interessati alla sola fase di riscaldamento.

### 3.4 VELOCIZZAZIONE DELLA PROCEDURA HTC\_PER E HTC\_WWW

Il coefficiente di permeabilità nella formulazione proposta da Bazant è per temperature minori uguali a 95°C si impiega una correlazione (di seguito riportata) funzione sia della temperatura che dell'umidità relativa, e per temperature maggiori di 95°C una correlazione dipendente solo dalla temperatura [1] .

$$a = a_o f_1(h) f_2(T) \quad \text{per } T \leq 95^\circ C$$

$$a = a_o f_2(95^\circ C) f_3(T) \quad \text{per } T > 95^\circ C$$

Inoltre per  $h > 1$   $f_1$  viene posto pari a 1. Operando con il linguaggio <gibiane> a livello di campi per punti ( cioè su strutture dati che si appoggiano ai nodi della magliatura ), non è possibile realizzare, direttamente, gli -if- per la selezione della correlazione da impiegare all'atto del calcolo del coefficiente di permeabilità nodo per nodo.

Occorre valutare, quindi le tre correlazione  $f_1$   $f_2$  e  $f_3$  anche in punti dove non dovrebbero essere applicate e poi ricombinare correttamente i vari contributi, tramite degli opportuni filtri operati utilizzando l'operatore MASQUE. Per ridurre significativamente il calcolo del coefficiente di permeabilità si è riscritta la procedura HTC\_PER adottando la seguente strategia. Si testa se la temperatura è in tutto il dominio minore di 95°C e si calcolano così solo  $f_1$  e  $f_2$  ; si testa se la temperatura è ovunque maggiore di 95°C e si calcola la  $f_3$ ; se si hanno temperature maggiori e minori di 95°C (caso misto) si calcolano tutte e tre le funzioni e si opera come sviluppato nella precedente versione. Tenuto conto che all'avvio del calcolo si richiedono in genere passi temporali ridotti la modifica ha comportato un significativo risparmio di tempo macchina.

Nella procedura HTC\_WWW si valuta il contenuto di acqua libera presente nel calcestruzzo secondo lo schema previsto da Bazant. Rimandando per i dettagli al documento [1] e [3] , è sufficiente qui ricordare che Bazant adotta tre correlazioni distinte selezionabili in base dell'umidità relativa: una per umidità inferiori a 0.93 (coso così detto insaturo), una per umidità maggiori di 1.06 (caso saturo) ed una zona con umidità compresa tra i due precedenti valori che funga da zona di raccordo.

Al momento sono state individuate due possibili modi di velocizzazione. Il primo simile a quanto fatto per HTC\_PER, cioè verificando che in tutto il dominio l'umidità è ovunque < di 0.93 e nel caso siano presenti punti sopra e sotto tale valore calcolare in contemporanea tutte e tre le correlazioni e quindi ricomporre il tutto con i filtraggi analogamente a quanto precedentemente sviluppato. Il secondo modo, sicuramente più efficace, di velocizzare la procedura è quello di operare a livello di singolo punto, ma ciò richiede uno sviluppo non più in <gibiane> ma direttamente in FORTRAN+ESOPE modificando i sorgenti di alcuni operatori di CAST3M.

Al momento tale operazione non è fattibile in ENEA, richiedendo sia il coinvolgimento di personale CEA che la disponibilità di pre-compileri ESOPE-FORTRAN. Il lavoro specifico sarà quindi oggetto delle attività previste per la fase conclusiva dell'accordo di collaborazione tra CEA ed ENEA su CAST3M per il prossimo anno.

### 3.5 ESTENSIONE A DOMINI TRIDIMENSIONALI

L'estensione a domini tridimensionali ha comportato un intervento limitato. E' stato necessario modificare il calcolo del prodotto scalare  $\text{grad}P \cdot \text{grad}T$  nell'equazione di bilancio del calore, il calcolo del calore asportato dal fluido che evapora dalla superficie esterna Figura 2, aggiungere la terza componente al coseno direttore delle superfici che scambiano massa con l'ambiente esterno e modificare l'inserimento e il controllo dei parametri di input alla procedura.




Figura 2 Asportazione di calore per evaporazione dell'acqua dalle superfici del calcestruzzo.

Va considerato però che i calcoli 3D, se inevitabili appesantiscono le analisi sia in termini di tempo macchina che d'ingombro di memoria e quindi si richiede che gli algoritmi siano ottimizzati quanto più possibile per rendere effettivamente fattibili tali analisi.

### 3.6 MIGLIORAMENTI FUNZIONALI

Data la forte non linearità del problema trattato, la messa a punto di un'analisi può risultare difficoltosa, soprattutto l'attraversamento della zona di transizione insaturo saturo può presentare instabilità numeriche con esplosione del calcolo. Occorre pertanto in presenza di tali effetti, monitorare più dettagliatamente l'andamento del calcolo, in particolare l'errore sul bilancio di massa e quindi operare un restart con adeguamento del passo di calcolo. Si è pertanto introdotta la possibilità di variare il livello delle stampe di controllo e di valutare graficamente le distribuzioni di grandezze critiche che segnalano l'inizio di una sofferenza che porta alla distruzione del processo iterativo, contenendo contemporaneamente le attività di I/O.

## 4 TEST DI CONVALIDA

Sono stati eseguiti un gran numero di test in modo da sollecitare tutte le parti modificate. Se ne riportano qui i due più significativi. Uno riguarda il riscaldamento iniziale di un elemento di un modulo di accumulo termico analizzato in assialsimmetria, l'altro il riscaldamento di un provino cubico modellizzato in 3D.

#### 4.1 MODULO DI ACCUMULO

Si è analizzato il primo riscaldamento di un modulo per accumulo termico. All'avvio del transitorio il provino è in equilibrio termo igrometrico con l'ambiente esterno, la temperatura è stata posta a 25°C e l'umidità relativa a 53%, a cui corrisponde una pressione parziale del vapore di 1700 Pa.

Il centro del cilindro si suppone essere un tubo in acciaio attraversato da un fluido a 300°C con un coefficiente di scambio pari a 100 W/m<sup>2</sup>-C; vista la ridotta resistenza termica del tubo rispetto non è stato necessario modellizzarlo, la condizione di scambio termico è stata quindi posta direttamente sul calcestruzzo. Il calcestruzzo scambia poi calore e massa verso l'ambiente esterno sempre in modo convettivo.

Nella Figura 3 sono riportati le caratteristiche geometriche e le condizioni di scambio applicate. Tale analisi risulta particolarmente gravosa in quanto il vapore non può fuoriuscire dal lato scaldato, ma necessariamente deve migrare radialmente verso il lato esterno posto in aria, sono quindi attese pressioni più elevate.




Figura 3 Schematico del modulo di accumulo con le condizioni al contorno termo-igrometriche.

La proprietà termoigrometriche del calcestruzzo sono corrispondenti a quelli di un calcestruzzo C60, per i valori si rimanda all'input posto in allegato.

Il transitorio si protratto per 4000 sec con 29700 passi di calcolo, valori e numero dei passi temporali sono riportati nella tabella seguente.

Numero passi	Valore passo (sec)	Tempo cumulato (sec)
1000	0.1	100
500	0.2	200
3200	0.25	1000
20000	0.1	3000
5000	0.2	4000

I risultati delle analisi sono costituiti ai vari istanti da distribuzione di temperatura, pressione, contenuto di acqua libera, acqua legata rilasciata e umidità relativa; tali dati vengono memorizzati da CAST3M su un file con estensione '.sauve' per essere poi rilette post-trattate e graficate sempre utilizzando CAST3M.

Di seguito sono riportati gli andamenti sopra menzionati in direzione radiale, cioè dalla faccia interna del provino scaldato alla faccia esterna, involuendo tali distribuzioni nel tempo.

Nella Figura 4 si riporta la distribuzione delle temperature. Si nota la progressiva diminuzione della pendenza iniziale e quindi del flusso di calore entrante nel calcestruzzo dovuto all'avvicinamento della temperatura di parete alla temperatura del fluido; va inoltre notata la difficoltà di caricamento del calore nella parte più esterna del modulo a seguito della bassa conducibilità del calcestruzzo.




Figura 4 Distribuzione radiale temperature in °C a vari istanti.

Correttamente la pressione assume pendenza nulla sul lato scaldato, essendo adiabatico rispetto al trasporto di massa; si nota inoltre che il rateo di crescita della pressione tenda a diminuire nel tempo a seguito della migrazione dell'acqua libera verso l'esterno.




Figura 5 Distribuzione radiale pressione del vapore in MPa a vari istanti.




Figura 6 Distribuzione radiale contenuto acqua libera in Kg/m3 a vari istanti.

La distribuzione dell'acqua libera Figura 6 mostra che il picco tende a diminuire a partire da un certo istante e ciò è congruente con il rallentamento nella salita della pressione. Si riscontra una perturbazione sul lato esterno conseguente ad una

densità di maglia più grossolana. L'umidità inoltre nella zona del picco supera il valore 1 e ciò indica che in quella zona si è in presenza di fluido saturo. Tale zona condiziona pesantemente la convergenza; per cui è stato necessario ripetere diverse volte le analisi al fine di individuare il corretto passo temporale da impiegare.




Figura 7 Distribuzione radiale dell'umidità relativa a vari istanti.

Andamento correlato al precedente si riscontra nel grafico dell'umidità relativa riportato in Figura 7. Il brusco cambiamento di pendenza è dovuto al forte cambio del coefficiente di permeabilità presente nelle correlazioni sviluppate da Bazant intorno ai 95°C.

Infine in Figura 8 si ha la distribuzione radiale del rilascio dell'acqua legata; si ricorda che il rilascio di acqua legata avviene quando si superano i 105°C ed è un processo irreversibile.




Figura 8 Distribuzione radiale della perdita di acqua legata Kg/m<sup>3</sup> a vari istanti.

Mediamente ad ogni passo di calcolo la convergenza è stata raggiunta dopo 8 iterazioni con errori relativi per la temperatura, la pressione e il bilancio di massa che risultano dell'ordine di 1.e-6. Il mantenimento di un errore contenuto sul bilancio di massa garantisce dall'esplosione del processo iterativo. In allegato sono riportati i file d'input e di post trattamento del caso.

## 4.2 PROVINO CUBICO TRIDIMENSIONALE

La discretizzazione del provino viste le simmetrie geometriche e dei carichi applicati è stata ridotta ad un sedicesimo come da figura sotto riportata. Sono stati impiegati elementi prismatici parabolici a 15 nodi, operando adeguati infittimenti nelle zone di spigolo che risultano più sollecitate da un punto di vista termo-igrometrico (Figura 9).

L'analisi è risultata particolarmente onerosa (un giorno e mezzo su HP con processore Intel Core-I5 con s.o. Windows 7), pertanto si è limitati ad un transitorio di 100 secondi. Il passo temporale impiegato è di 0.02 secondi, per un totale di 5000 passi di calcolo. Per rendere fattibili analisi 3D occorrerebbe velocizzare, come già detto, la procedura per il calcolo del contenuto dell'acqua legata. Occorrerebbe inoltre esplorare la possibilità di impiego, su idonea macchina, delle opzioni per il calcolo parallelo presenti in CAST3M. Su tutta la superficie esterna è applicata la condizione di scambio termico ed igrometrico convettiva. In questo caso si ha scambio di massa sullo stesso lato che viene scaldato.






Figura 9 Magliatura test provino cubico con dettaglio zona spigolo.




Figura 10 Distribuzione temperature (°C) zona spigolo dopo 100 secondi.




Figura 11 Distribuzione pressione (MPa) zona spigolo dopo 100 secondi.




Figura 12 Distribuzione contenuto acqua libera (Kg/m3) zona spigolo dopo 100 secondi.




Figura 13 Distribuzione umidità relativa zona spigolo dopo 100 secondi.

In si riporta la distribuzione della temperatura nella zona dello spigolo dopo 100 sec di transitorio. La temperatura massima raggiunta è di 72 °C, sotto il valore di soglia della deidratazione. La distribuzione risulta congruente con le condizioni al contorno imposte. Il si ha la distribuzione delle pressioni sempre nella zona dello spigolo e a 100 sec. Si notano qui però alcune anomalie collegate alla discretizzazione della maglia, che non consente di modellizzare completamente la distribuzione quando il massimo di pressione è concentrato nel primo strato di elementi. In particolare si nota dopo la zona del picco di pressione una zona di minimo e poi una risalita della pressione. Tale inconveniente si potrebbe ridurre infittendo maggiormente la maglia con aggravio però dei tempi di calcolo. Si ritiene comunque, in base ad esperienza acquisita su analoghe condizioni di carico e su geometrie monodimensionali meno onerose, che ciò non alteri significativamente le distribuzioni nei successivi istanti del transitorio.

Tali anomalie si riscontrano inoltre negli andamenti del contenuto d'acqua legata e dell'umidità relativa (rispettivamente Figura 12 e Figura 13), essendo esse grandezze funzioni della pressione e della temperatura. Si notano inoltre asimmetrie dovute all'orientazione degli elementi prismatici, che come gli elementi triangolari possono dare comportamenti anisotropi. Ma tutto ciò rientra nei normali comportamenti dei metodi f.e.m. e non è imputabile agli sviluppi effettuati.

## 5 CONCLUSIONI

Un sensibile miglioramento qualitativo e quantitativo è stato apportato alla precedente versione di HTCTRAN. Test selettivi sono stati effettuati per controllare le parti modificate. Sono stati inoltre congiuntamente a personale CEA individuate ulteriori linee di sviluppo che riguardano essenzialmente l'accoppiamento con il

 <b>Ricerca Sistema Elettrico</b>	<b>Sigla di identificazione</b>	<b>Rev.</b>	<b>Distrib.</b>	<b>Pag.</b>	<b>di</b>
	NNFISS – LP1 - 019	0	L	18	59

modulo meccanico esistente in CAST3M. Al momento l'azione sembra piuttosto onerosa vista la complessità della procedura per il calcolo meccanico PASAPAS. Altra soluzione potrebbe essere di costruire un modulo meccanico più mirato alla termo-meccanica del calcestruzzo, indipendente dalla PASAPAS; un primo accoppiamento della parte termo-igrometrica con un modulo meccanico elastico potrebbe essere però realizzato rapidamente, consentendo la messa a punto dei carichi meccanici dipendenti dalla termo-igrometria.

Gli sviluppi effettuati possono quindi essere introdotti nella versione corrente di CAST3M e conseguentemente aggiornati i manuali d'uso.

## 6 ALLEGATO 1 :LISTATO DELLA PROCEDURA HTCTRAN E PROCEDURE AD ESSA COLLEGATE

Nota : i caratteri in rosso evidenziano le parti modificate e aggiunte

```

$$$$ HTCTRAN
* HTCTRAN  PROCEDUR  AM          09/12/15   21:15:17   6588
*-----
*23456789012345678901234567890123456789012345678901234567890123456789012
*-----
*-----
'DEBPROC' HTCTRAN  N_PAS*'ENTIER'  NN_PAS*'ENTIER'  N_ITER*'ENTIER'
              T_PAS*'FLOTTANT'  MODL1*'MMODEL'  TAV1*'TABLE';
*-----
'MESS' '-----';
'MESS' '-----HTCTRAN-----';
'MESS' '-----';
'MESS' '---Bazant coupled moisture and heat transfer in concrete-----';
'MESS' '-----';
'MESS' ' Conservation of mass -----';
'MESS' '-----';
'MESS' '      d w      dwd              a -----';
'MESS' '      --- - --- = - div J , J= - --- grad P -----';
'MESS' '      d t      d t              g -----';
'MESS' '-----';
'MESS' ' Heat balance -----';
'MESS' '-----';
'MESS' '      d T      d w -----';
'MESS' ' (De Cp)--- - Ca --- - Cw J grad T = - div q , q= - k grad T -----';
'MESS' '      d t      d t -----';
'MESS' '-----';
'MESS' '---last mod. - : 25-07-11 - hour :-15.00-----';
'MESS' '-----';
*-----
* Auteur   : G.M. Giannuzzi, ENEA-UTRINN-PCI   (1998)
*-----
* Modifié  : G.M. Giannuzzi, ENEA-UTRINN-PCI le 25-07-2011
*-----
*
* DESCRIPTION:  Procedure for heat and moisture
*              transfer analysis in concrete based
*              on a modified Bazant model.
*              Axisymmetric and plane cases and 3D
*              with several boundaries

```

\* exchanging vapour and heat.

\*

\*-----

\*

'SI' ('NON' ('EXIS' TAV1 'DETR' ));

FDET = VRAI; TAV1.'DETR'= FDET;

'SINON';

FDET=TAV1.'DETR';

'FINSI';

'SI' ('NON' ('EXIS' TAV1 'GRAF' ));

GRAF = FAUX; TAV1.'GRAF'= GRAF;

'SINON';

GRAF=TAV1.'GRAF';

'FINSI';

'SI' ('NON' ('EXIS' TAV1 'FLG1' ));

FLG1 = FAUX; TAV1.'FLG1'= FLG1;

'SINON';

FLG1=TAV1.'FLG1';

'FINSI';

'SI' ('NON' ('EXIS' TAV1 'FLG' ));

FLG = 0; TAV1.'FLG'= FLG;

'SINON';

FLG=TAV1.'FLG';

'FINSI';

\*

\* Mesh and model definition

\*

MAGLIA='EXTR' MODL1 'MAIL';

\*

'SI' ('EGA' ('VALE' 'MODE') 'AXIS');

FF1= 'MOT' 'T,R';

FF2= 'MOT' 'T,Z';

FF3= 'MOT' ' ';

LICF='MOTS' FF1 FF2;

LICJ='MOTS' 'JX\_R' 'JY\_Z';

'SINON' ;

'SI' (('EGA' ('VALE' 'MODE') 'PLANDEFO') 'OU'

('EGA' ('VALE' 'MODE') 'PLANCONT'));

FF1= 'MOT' 'T,X';

FF2= 'MOT' 'T,Y';

FF3= 'MOT' ' ';

LICF='MOTS' FF1 FF2;

LICJ='MOTS' 'JX\_R' 'JY\_Z';

'SINON';

'SI' ('EGA' ('VALE' 'MODE') 'TRID');

FF1= 'MOT' 'T,X';

FF2= 'MOT' 'T,Y';

```

FF3= 'MOT' 'T,Z';
LICF='MOTS' FF1 FF2 FF3;
LICJ='MOTS' 'JX_R' 'JY_Z' 'JZ_Z';
'SINON';
'MESS' '*** ERREUR - MODELE INCOMPATIBLE ' ;
'QUITTER' HTCTRAN;
'FINSI';
'FINSI';
'FINSI';
*
LIPT='MOTS' 'T';
LICP='MOTS' 'P' 'LX' ;
LICT='MOTS' 'T' 'LX' ;
*
*=====
* ANALISYS OF THE DATA SUPPLIED IN INPUT TABLE
*=====
*
*PARAMETRI DI CONTROLLO
*
'SI'('NON'('EXIS' TAV1 'GAMMA'));
GAMMA=1.;TAV1.'GAMMA'=GAMMA;
'SINON';
GAMMA=TAV1.'GAMMA';
'FINSI';
*
* VARIAZIONE DI P E T
* PER CUI SCATTA LA DERIVATA TANGENTE
*
'SI'('NON'('EXIS' TAV1 'EPSILON' ));
EPS1=1.E-8;TAV1.'EPSILON'=EPS1;
'SINON';
EPS1=TAV1.'EPSILON';
'FINSI';
*
* ERRORI SULLA CONVERGENZA
*
'SI'('NON'('EXIS' TAV1 'ERPM' ));
ERPM=1.E-4;TAV1.'ERPM'=ERPM;
'SINON';
ERPM=TAV1.'ERPM';
'FINSI';
'SI'('NON'('EXIS' TAV1 'ERTM' ));
ERTM=1.E-4;TAV1.'ERTM'=ERTM;
'SINON';
ERTM=TAV1.'ERTM';
'FINSI';

```

 <b>Ricerca Sistema Elettrico</b>	<b>Sigla di identificazione</b>	<b>Rev.</b>	<b>Distrib.</b>	<b>Pag.</b>	<b>di</b>
	NNFISS – LP1 - 019	0	L	22	59

```

*
* Crank-Nicholson's scheme: time integration constant
*
'SI' ( 'NON' ( 'EXIST' TAV1 'LAMBDA' ));
  LAMBDA=0.5;TAV1 . 'LAMBDA'= LAMBDA;
'SINON';
  LAMBDA= TAV1 . 'LAMBDA';
'FINSI';
  UMLAMBDA=1.-LAMBDA;
*
* Reference temperature [°C ]
*
'SI' ( 'NON' ( 'EXIST' TAV1 'TEMPERATURE_INITIALE' ));
  TREF=25.;TAV1 . 'TEMPERATURE_INITIALE'= TREF;
'SINON';
  TREF = TAV1 . 'TEMPERATURE_INITIALE' ;
'FINSI';
*
* Reference pressure [N/mm²= Mpa]
*
'SI' ( 'NON' ( 'EXIST' TAV1 'PRESSION_INITIALE' ));
  'MESS' '*** ERREUR - IL MANQUE LA PRESSION INITIALE' ;
  'QUITTER' HTCTRAN;
'SINON';
  PREF = TAV1 . 'PRESSION_INITIALE' ;
'FINSI';
*
* water saturation content at 25 °C [Kg/mc]
*
'SI' ( 'NON' ( 'EXIST' TAV1 'w1' ));
  TAV1.'w1'=100.;
'FINSI';
*
* Dehydration water specific heat [J/Kg]
*
'SI' ( 'NON' ( 'EXIST' TAV1 'CAD' ));
  CAD=0.2328E6;
  TAV1.'CAD'= CAD;
'SINON';
  CAD=TAV1.'CAD';
'FINSI';
*Heat capacity of interstitial and bound water J/kg-C
  CBW=3760.;
*
* Smooth for Ca as a function of h
*
*PL1 = 'PROG' 0. 95. 95.5 96. 96.5 97. 98. 99. 100. 101. 102.

```



```

*          103. 103.5  104. 104.5  105. 1000. ;
*PL2 = 'PROG' 0. 0. 0.0062 0.0245 0.0545 0.0955 0.2061 0.3455 0.5
*          0.6545 0.7939 0.9045 0.9455 0.9755 0.9938  1.  1. ;
*
PL1 = 'PROG' 0. 1. 'PAS' 0.005 1.06 20. ;
PL2 = 'PROG' 1. 1. 0. 0. ;
'REPE' LOOPA (( 'DIME' PL1) - 4) ;
  I = &LOOPA + 2 ;
  VPL1 = 'EXTR' PL1 I ;
  VPL2 = ('COS'((VPL1 - 1.)/0.06*90.))**2;
  PL2 = 'INSE' PL2 I VPL2 ;
'FIN' LOOPA;
EVCA = 'EVOL' 'MANU' PL1 'SCAL' PL2 'SCAL' ;
*
*  Cement content [Kg/mc]
*
'SI' ( 'NON' ( 'EXIST' TAV1 'C' ));
  CC=300.;TAV1.'C'=CC;
'SINON';
  CC=TAV1.'C';
'FINSI';
*
*  Initial permeability [m/s]
*
'SI' ( 'NON' ( 'EXIST' TAV1 'A0' ));
  TAV1.'A0'=1.E-13;
'FINSI';
*
*  Thermal capacity of the drying concrete [J/(m3*C)]
*
*
*  Density of dry concrete a 25 °C Kg/m3
*
'SI' ( 'NON' ( 'EXIST' TAV1 'DEN_SEC' ));
  DEN_SEC= 2400.; TAV1 . 'DEN_SEC' = DEN_SEC;
'SINON';
  DEN_SEC=TAV1 . 'DEN_SEC';
'FINSI';
*
*  Specific heat of dry concrete a 25 °C J/kg-C
*
'SI' ( 'NON' ( 'EXIST' TAV1 'CCP_SEC' ));
  CCP_SEC= 880.; TAV1 . 'CCP_SEC' = CCP_SEC;
'SINON';
  CCP_SEC=TAV1 . 'DEN_SEC';
'FINSI';
  ROCSEC=CCP_SEC * DEN_SEC;

```

```

TAV1 . 'ROCSEC' = ROCSEC;
*
* Thermal conductivity of the concrete [W/m*K]
*
'SI' ( 'NON' ( 'EXIST' TAV1 'K0_SEC' ));
  K0_SEC =1.92 ; TAV1 . 'K0_SEC' = K0_SEC;
  K1_SEC =-0.00125; TAV1 . 'K1_SEC' = K1_SEC;
'SINON';
  K0_SEC= TAV1 . 'K0_SEC';
  K1_SEC= TAV1 . 'K1_SEC';
'FINSI';
*
* volumetric elasticity modulus MPa
*
'SI' ( 'NON' ( 'EXIST' TAV1 'E0' ));
  E0=35000. ;TAV1.'E0'= E0 ;
'SINON';
  E0=TAV1.'E0' ;
'FINSI';
'SI' ( 'NON' ( 'EXIST' TAV1 'NU' ));
  NU=.18;TAV1.'NU'=NU;
'SINON';
  NU=TAV1.'NU' ;
'FINSI';
* ---> BM = E0/3(1-2ν) * E/E0(T)
'SI' ( 'NON' ( 'EXIST' TAV1 'EE0T' ));
* Due to DTU
  EE01 = 'PROG' 1. 1. .5 .15 .05 ;
  TTT1 = 'PROG' 0. 50. 200. 400. 600. ;
'SINON';
  EE0T= TAV1.'EE0T';
  TTT1='EXTR' EE0T 'ABSC';
  EE01='EXTR' EE0T 'ORDO';
'FINSI';
  EE01 =(EE01/(3.*(-2.* NU + 1.)))*E0;
  EBM ='EVOL' 'MANU' TTT1 EE01 ;
  TAV1.'EBM'=EBM;
*
* Coefficient of linear thermal dilatation[/ $\theta$ C]
*
'SI' ( 'NON' ( 'EXIST' TAV1 'ALFA' ));
  ALFA=9.E-6;TAV1.'ALFA'=ALFA;
'SINON';
  ALFA = TAV1.'ALFA';
'FINSI';
*
* Dehydration curve of concrete

```

```

*
'SI' ( 'NON' ( 'EXIST' TAV1 'EWD' ));
* PT = 'PROG' 0. 105. 180. 200. 220. 240. 260.
*      280. 300. 320. 340. 360. 400. 600. ;
* PWD = 0.66 * ('PROG' 0. 0. .264 .302 .498 1.057 3.358
*      5.283 6.377 7.019 7.283 7.472 7.660 8.075 )*(CC/100.);
'SI' ( 'NON' ( 'EXIST' TAV1 'F_STE'));
*Stoichiometric factor
  F_STE = 0.24; TAV1 . 'F_STE' = F_STE;
'SINON';
  F_STE=TAV1 .'F_STE';
'FINSI';
'SI' ( 'NON' ( 'EXIST' TAV1 'F_INV'));
*Aging factor
  F_INV = 0.95; TAV1 . 'F_INV' = F_INV;
'SINON';
  F_INV=TAV1 .'F_INV';
'FINSI';
PT1 = 'PROG' 105. 'PAS' 5. 800.;
PT = ('PROG' 0. ) 'ET' PT1;
N_PT1='DIME' PT1; PU='PROG' N_PT1*1.;
PWD=(PU+ (SIN ((180./2.)*
  (PU-(2.*(EXP (-0.004*(PT1-(105.*PU)))))))))/2.;
PWD= (F_STE*F_INV*CC)*(('PROG' 0.) 'ET' PWD);
EWD = 'EVOL' 'MANU' PT PWD ;TAV1.'EWD'=EWD;
'SINON';
  EWD=TAV1.'EWD';
  PWD='EXTR' EWD 'ORDO';
'FINSI';
*DESS EWD 'TITR' 'water released by dehydrazion Kg/m3';
WDO='MAXI' PWD;
*
* Thermal stiffness matrix
*
'SI' ( 'EXIST' TAV1 'CONDUCTIVITE_THERMIQUE' );
'MESS' ' CONDUCTIVITE THERMIQUE ' ;
  RIG_V = TAV1 . 'CONDUCTIVITE_THERMIQUE' ;
'FINSI';
*
* Thermal load
*
IFT =FAUX ;
IFTC =FAUX ;
'SI' ( 'EXIST' TAV1 'CONVECTION_THERMIQUE' );
'MESS' ' CONVECTION THERMIQUE ' ;
  TCHARCT = TAV1 . 'CONVECTION_THERMIQUE' ;
  TCONDC='EXTR' RIG_V 'RIGI' 'NOMU';

```

```

    IFT =VRAI ;
    IFTC =VRAI ;
'FINSI';
    IFTB =FAUX ;
'SI' ( 'EXIST' TAV1 'BLOCAGES_THERMIQUE' );
'MESS' ' BLOCAGES THERMIQUE ' ;
    TCHARTI = TAV1 . 'BLOCAGES_THERMIQUE' ;
    TCONDB='EXTR' RIG_V 'RIGI' 'MULT';
    IFT =VRAI ;
    IFTB =VRAI ;
'FINSI';
'SI' ( 'EXIST' TAV1 'FLUX_THERMIQUE' );
'MESS' ' FLUX THERMIQUE ' ;
    TCHARFT = TAV1 . 'FLUX_THERMIQUE' ;
    IFT = VRAI ;
'FINSI';
*
* Hydraulic stiffness matrix
*
'SI' ( 'EXIST' TAV1 'CONDUCTIVITE_HYDRAULIQUE' );
    MESS ' CONDUCTIVITE HYDRAULIQUE';
    PCOND2 = TAV1 . 'CONDUCTIVITE_HYDRAULIQUE' ;
'FINSI';
*
* Hydraulic load
*
    IFP = FAUX;
    IFPC = FAUX;
'SI' ( 'EXIST' TAV1 'CONVECTION_HYDRAULIQUE' );
'MESS' ' CONVECTION HYDRAULIQUE' ;
    PCHARCH = TAV1 . 'CONVECTION_HYDRAULIQUE' ;
    PCONDC='EXTR' PCOND2 'RIGI' 'NOMU';
    IFP = VRAI ;
    IFPC = VRAI ;
'FINSI';
    IFPB = FAUX ;
'SI' ( 'EXIST' TAV1 'BLOCAGES_HYDRAULIQUE' );
'MESS' ' BLOCAGES_HYDRAULIQUE' ;
    PCHARPI = TAV1 . 'BLOCAGES_HYDRAULIQUE' ;
    PCONDB='EXTR' PCOND2 'RIGI' 'MULT';
    IFP = VRAI ;
    IFPB = VRAI ;
'FINSI';
'SI' ( 'EXIST' TAV1 'FLUX_HYDRAULIQUE' );
'MESS' ' FLUX_HYDRAULIQUE ' ;
    PCHARFH = TAV1 . 'FLUX_HYDRAULIQUE' ;
    IFP = VRAI ;

```

```

'FINSI';
'SI' IFP;
'SI' ('EXIST' TAV1 'FRONTIERES_PRESSION');
  NFRONT= 'DIME' TAV1.'FRONTIERES_PRESSION';
'SINON';
'MESS' '*** ERREUR - DANS LES CONDITIONS AUX LIMITES'
      ' EN PRESSION ' ;
'FINSI';
'FINSI';
*-----
*
*           RESTART PROCEDURE
*-----
*
'SI' ( 'NON' ('EXIST' TAV1 'RESULTATS' ));
  RESTART1= FAUX;
  TEMPO = 0.;
  T0   = 'MANU' 'CHPO' MAGLIA 1 'SCAL' TREF ;
  P0   = 'MANU' 'CHPO' MAGLIA 1 'SCAL' PREF ;
  VT0  = 'MANU' 'CHPO' MAGLIA 1 'SCAL' 0. ;
  VP0  = 'MANU' 'CHPO' MAGLIA 1 'SCAL' 0. ;
  WI HI = HTC_WWW  T0      P0      TAV1 ;
  WT1 HT1 = HTC_WWW (T0 + 1.) P0      TAV1 ;
  DWT0    = WT1 - WI ;
  WP1 HP1 = HTC_WWW  T0      (P0 + 0.00001) TAV1 ;
  DWPO    = (WP1 - WI)/0.00001 ;
  DWD0    = 'MANU' 'CHPO' MAGLIA 1 'SCAL' 0.;
  WD      = 'MANU' 'CHPO' MAGLIA 1 'SCAL' 0.;
  TAV1.'WDM'= WD;
*
  CHI='CHAN' 'CHAM' MODL1 WI;
  MI ='INTG' MODL1 CHI;
  MT = MI ;
  PSAT = HTC_WTR TAV1 'SATU' T0 ;
*
  TAV1.'TMAX'=T0;
  T0='NOMC' 'T' T0;
  P0='NOMC' 'T' P0;
*
'MESS' '-----';
'MESS' ' MASSE D EAU INITIALE ' MI;
'MESS' ' H.R.          INITIALE ' ('MAXI' HI);
'MESS' ' PRE. SAT.    INITIALE ' ('MAXI' PSAT);
*
  PERMAS=0.;
  IPAS=0;
*
  TAV1.'MI'          = MI;

```

```

TAV1.'RESULTATS'      = 'TABLE';
TAV1.'RESULTATS'.IPAS = 'TABLE';
TAV1.'RESULTATS'.IPAS.'INSTANT'      = 0. ;
TAV1.'RESULTATS'.IPAS.'TEMPERATURE'  = T0 ;
TAV1.'RESULTATS'.IPAS.'PRESSION'     ='NOMC' 'P' P0;
TAV1.'RESULTATS'.IPAS.'HUMIDITE_RELATIVE' ='NOMC' 'HR' HI;
TAV1.'RESULTATS'.IPAS.'EAU_LIBRE'    ='NOMC' 'WF' WI;
TAV1.'RESULTATS'.IPAS.'EAU_LIEE'     ='NOMC' 'WD' WD;
TAV1.'RESULTATS'.IPAS.'MASSE_PERDUE'  = PERMAS;
* TAV1.'RESULTATS'.IPAS.'MASSE_FLUX'   = JFL ;
TAV1.'RESULTATS'.IPAS.'MASSE_BILAN'   = MT ;
TAV1.'RESULTATS'.IPAS.'MASSE_RELACHE' = 0. ;
TAV1.'RESULTATS'.IPAS.'PRES_SAT'     = 'NOMC' 'PSAT' PSAT;
TAV1.'VT0' = VT0 ;
TAV1.'VP0' = VP0 ;
TAV1.'DWD0' = DWD0 ;
TAV1.'DWT0' = DWT0 ;
TAV1.'DWPO' = DWPO ;
'SI' FDET;
'DETR' HI ; 'DETR' WI ; 'DETR' CHI;
'DETR' WT1 ; 'DETR' HT1 ;
'DETR' WP1 ; 'DETR' HP1 ;
'FINSI';
'SINON';
RESTART1 = VRAI ;
MI=TAV1.'MI';
NPSAUVE= 'DIME' (TAV1 . 'RESULTATS');
IPAS=NPSAUVE-1;
TEMPO = TAV1.'RESULTATS' . IPAS . INSTANT;
T0     = TAV1.'RESULTATS' . IPAS . 'TEMPERATURE';
P0     ='NOMC' LICP LICT TAV1.'RESULTATS'.IPAS.'PRESSION';
PERMAS = TAV1 . 'RESULTATS' . IPAS . 'MASSE_PERDUE';
MT     = TAV1 . 'RESULTATS' . IPAS . 'MASSE_BILAN' ;
MD     = TAV1 . 'RESULTATS' . IPAS . 'MASSE_RELACHE';
'FINSI';
*
USDT = 1./T_PAS ;
*
II = 1 ;
N1_PAS= 0;
*
*=====
* BOUCLE SUR LES PAS DE TEMPS
*=====
*
'REPE' TOTO N_PAS;
*

```

```

  Istant0 = ((II - 1)*T_PAS )+ TEMPO ;
  Istant1 = ( II      *T_PAS )+ TEMPO ;
  ITER = 0 ;
*
'MESS' ' ' ;
'MESS' '-----';
'MESS' '** PAS  ' II ' INSTANT ' Istant1 ' **';
'MESS' '-----';
*
*TOS  temp. comp. scal e senza lagr. passo pre.
*POS  pres. comp. scal e senza lagr. passo pre.
*T1S  temp. comp. scal e senza lagr. passo att. iter. pre.
*P1S  pres. comp. scal e senza lagr. passo att. iter. pre.
*VT0  velocita temp. iter. o passo pre.
*VPO  velocita pres. iter. o passo pre.
*
VT0 =TAV1.'VT0';
VPO =TAV1.'VPO';
DWD0=TAV1.'DWD0';
DWT0=TAV1.'DWT0';
DWPO=TAV1.'DWPO';
*
TOS = 'EXCO' 'T' T0 'SCAL';
POS = 'EXCO' 'T' P0 'SCAL';
T1S = TOS+(T_PAS*GAMMA*VT0);
P1S = POS+(T_PAS*GAMMA*VPO);
*
IstantM = (UMLAMBDA*Istant0) + (LAMBDA*Istant1);
*
* Thermal Loads
*
'SI' IFT ;
  TCONV = 0.;
'SI' ('EXIST' TAV1 'CONVECTION_THERMIQUE' );
  TCNVCT = 'TIRE' TCHARCT IstantM ;
  TCONV = TCONV + TCNVCT ;
'FINSI';
'SI' ('EXIST' TAV1 'BLOCAGES_THERMIQUE' );
*  TCNVTI = 'TIRE' TCHARTI Istant1 ;
  TCNVTI = 'TIRE' TCHARTI IstantM ;
  TCONV = TCONV + TCNVTI ;
'FINSI';
'SI' ('EXIST' TAV1 'FLUX_THERMIQUE' );
  TCNVFT = 'TIRE' TCHARFT IstantM ;
  TCONV = TCONV + TCNVFT ;
'FINSI';
'FINSI';

```

```

*
* Hydraulic Loads
*
'SI' IFP ;
  PCONV = 0. ;
'SI' ('EXIST' TAV1 'CONVECTION_HYDRAULIQUE' );
  PCNVCH = 'TIRE' PCHARCH Istantm ;
  PCONV = PCONV + PCNVCH;
'FINSI';
'SI' ('EXIST' TAV1 'BLOCAGES_HYDRAULIQUE' );
* PCNVPI = 'TIRE' PCHARPI Istant1 ;
  PCNVPI = 'TIRE' PCHARPI Istantm ;
  PCONV =PCONV + PCNVPI;
'FINSI';
'SI' ('EXIST' TAV1 'FLUX_HYDRAULIQUE' );
  PCNVFH = 'TIRE' PCHARFH Istantm ;
  PCONV =PCONV + PCNVFH;
'FINSI';
'FINSI';
IFITER = FAUX ;
*=====
*           ITERATIONS
*=====
'REPE' OTO ( N_ITER + 1 );
*
  ITER= ITER + 1;
'MESS' '** ITERATION ' ITER ' **';
*
*-----
* Specific Water Content of Concrete
* Calculation of the derivatives
* dw/dT and dw/dP
* and Relative Humidity
*-----
*
DT1A = T1S-T0S ;
DP1A = P1S-POS ;
DT1Z = 'MASQUE' ('ABS' DT1A) 'EGINFE' EPS1;
DP1Z = 'MASQUE' ('ABS' DP1A) 'EGINFE' EPS1;
DT1Z1 = 'MASQUE' (1. - DT1Z) 'SUPERIEUR' .9;
DP1Z1 = 'MASQUE' (1. - DP1Z) 'SUPERIEUR' .9;
DT1 = DT1A + DT1Z;
DP1 = DP1A + DP1Z;
*
'SI' FDET ;
'DETR' DT1A ; 'DETR' DP1A ;
'FINSI';

```



```

*
'SI' ('EGA' ITER 1);
  WT0P0 H00 = HTC_WWW  T0S P0S TAV1 ;
'FINSI';
  WT1P0 H10 = HTC_WWW  T1S P0S TAV1 ;
  WT1P1 H11 = HTC_WWW  T1S P1S TAV1 ;
  WT0P1 H01 = HTC_WWW  T0S P1S TAV1 ;
*----> dw/dT
  DWTM=(WT1P0-WT0P0+WT1P1-WT0P1)/2.;
  DWT =(DWTM / DT1 * DT1Z1) + (DT1Z1*DWT0);
*----> dw/dP
  DWPM=(WT0P1-WT0P0+WT1P1-WT1P0)/2.;
  DWP =(DWPM / DP1 * DP1Z1) + (DP1Z1*DWP0);
*----> W HR
  PPP=(UMLAMBDA*P0S)+(LAMBDA*P1S);
  TTT=(UMLAMBDA*T0S)+(LAMBDA*T1S);
  WWW HHH = HTC_WWW  TTT PPP TAV1 ;
*
'SI' FDET ;
'DETR' H11 ;
'DETR' H10 ; 'DETR' H01 ;
'DETR' WT1P0; 'DETR' WT0P1 ;
'DETR' WT1P1;
'FINSI';
*
'SI' (FLG '>EG' 4);
'MESS' '-----'
      '-----';
'MESS' 'DWDT-MAX' ('MAXI' DWT ) 'DWDT-MIN' ('MINI' DWT )
      'DWDP-MAX' ('MAXI' DWP ) 'DWDP-MIN' ('MINI' DWP );
'MESS' '-----'
      '-----';
'FINSI';
*
'SI' GRAF;
* titr 'hhh in 'ISTANT1 ITER;
* trac maglia HHH;
* titr 'www in 'ISTANT1 ITER;
* trac maglia WWW;
* titr 'T in 'ISTANT1 ITER;
* trac maglia TTT;
* titr 'P in 'ISTANT1 ITER;
* trac maglia PPP;
  titr ' dwp in'ISTANT1 ITER;
  trac maglia DWP;
  titr ' dwt in'ISTANT1 ITER;
  trac maglia DWT;

```

```

'FINSI';
*
*-----
* Chemically Bound Water and Derivative
*-----
*
      WD DF1= HTC_CHBW T1S HHH TAV1 ;
      DWD   =(DF1 / DT1 * DT1Z1) + (DT1Z * DWD0);
*
*-----
* Permeability
*-----
*
      AGDG1=HTC_PER TTT HHH TAV1 ;
      AGDG = 1.E6 * AGDG1 ;
'SI' GRAF;
      titr 'Permeability' Istant1 ITER;
      trac AGDG maglia;
'FINSI';
*
'SI' (FLG '>EG' 5);
'MESS' 'RANGES OF RELATIVE HUMIDITY, WATER CONTENT, PERMEABILITY';
'MESS' 'H -MAXI ' ('MAXI' HHH) 'H -MINI ' ('MINI' HHH);
'MESS' 'W -MAXI ' ('MAXI' WWW) 'W -MINI ' ('MINI' WWW);
'MESS' 'A -MAXI ' ('MAXI' AGDG) 'A -MINI ' ('MINI' AGDG);
'FINSI';
*
*-----
* Specific heat of water
*-----
*
      CW= HTC_WTR TAV1 'CW' TTT;
*
*-----
* Heat of water vaporization
*-----
*
      CA= HTC_WTR TAV1 'HFG' TTT;
*
*-----
* Heat convection term
*-----
      TTTT ='NOMC' 'T' TTT;
      PPPP ='NOMC' 'T' PPP;
      GRAP='GRAD' MODL1 PPPP;
      GRAT='GRAD' MODL1 TTTT;
      GRATX='MANU' 'CHML' MODL1 'SCAL' ('EXCO' FF1 GRAT 'SCAL') RIGIDITE;

```

```

GRATY='MANU' 'CHML' MODL1 'SCAL' ('EXCO' FF2 GRAT 'SCAL') RIGIDITE;
GRAPX='MANU' 'CHML' MODL1 'SCAL' ('EXCO' FF1 GRAP 'SCAL') RIGIDITE;
GRAPY='MANU' 'CHML' MODL1 'SCAL' ('EXCO' FF2 GRAP 'SCAL') RIGIDITE;
ZZZZ1 = 'CHAN' 'NOEUD' MODL1 GRAPX;
GRAPCX = 'CHAN' 'CHPO' MODL1 ZZZZ1;
ZZZZ2 = 'CHAN' 'NOEUD' MODL1 GRAPY;
GRAPCY = 'CHAN' 'CHPO' MODL1 ZZZZ2;
'SI' ('NEG' FF3 ' ');
  GRATZ='MANU' 'CHML' MODL1 'SCAL' ('EXCO' FF3 GRAT 'SCAL') RIGIDITE;
  GRAPZ='MANU' 'CHML' MODL1 'SCAL' ('EXCO' FF3 GRAP 'SCAL') RIGIDITE;
  ZZZZ4 = 'CHAN' 'NOEUD' MODL1 GRAPZ;
  GRAPCZ = 'CHAN' 'CHPO' MODL1 ZZZZ4;
'FINSI';
LM1= 'MOTS' 'SCAL' 'SCAL';
LM2= 'MOTS' 'SCAL' 'SCAL';
LM3= 'MOTS' 'SCAL' 'SCAL';
GRAPTX = MODL1 GRAPX * GRATX LM1 LM2 LM3;
GRAPTY = MODL1 GRAPY * GRATY LM1 LM2 LM3;
GRAPTXY = GRAPTX + GRAPTY;
'SI' ('NEG' FF3 ' ');
  GRAPTZ = MODL1 GRAPZ * GRATZ LM1 LM2 LM3;
  GRAPTXY = GRAPTXY + GRAPTZ;
'FINSI';
ZZZZ3 = 'CHAN' 'NOEUD' MODL1 GRAPTXY;
GRAPTCXY = 'CHAN' 'CHPO' MODL1 ZZZZ3 ;
*----> - cw * a/g * grad p * grad t
AGPGT = 'SOURCE' MODL1 (-1. * CW * AGDG * GRAPTCXY);
'SI' FDET;
'DETR' PPPP ; 'DETR' TTTT ;
'DETR' GRAP ; 'DETR' GRAT ;
'DETR' GRATX ; 'DETR' GRATY ;
'DETR' GRAPTX ; 'DETR' GRAPTY ;
'DETR' GRAPTXY; 'DETR' GRAPTCXY;
'DETR' ZZZZ1 ; 'DETR' ZZZZ2 ; 'DETR' ZZZZ3;
'SI' ('NEG' FF3 ' ');
'DETR' GRATZ ; 'DETR' GRAPTZ ; 'DETR' ZZZZ4;
'FINSI';
'FINSI';
*-----
* Water Evaporation Heat on various frontiers
* Ca*J*n
*-----
'SI' IFP;
IFR=0;
CAJN_T = 0.;
'REPE' BOUC1 NFRONT;
IFR=IFR + 1;

```

```

COSXR =TAV1.'FRONTIERES_PRESSION'. IFR .'CODIRXR' ;
COSYZ =TAV1.'FRONTIERES_PRESSION'. IFR .'CODIRYZ' ;
MAI_FP=TAV1.'FRONTIERES_PRESSION'. IFR .'MAILLAGE';
'SI' ('NEG' FF3 ' ');
COSZZ=TAV1.'FRONTIERES_PRESSION'. IFR .'CODIRZZ' ;
CAJN ='REDU'(1.*AGDG*CA*((GRAPCX*COSXR)+(GRAPCY*COSYZ)
+(GRAPCZ*COSZZ))) MAI_FP;
'SINON';
CAJN ='REDU'(1.*AGDG*CA*((GRAPCX*COSXR)+(GRAPCY*COSYZ))) MAI_FP;
'FINSI';
FCAJN='FLUX' MODL1 CAJN ;
CAJN_T = CAJN_T + FCAJN ;
'FIN' BOUC1;
'FINSI';
*
*---> Ca*dw/dP*dP/dt
*
*list hhh;
SMCA = 'NOMC' 'SCAL' ('VARI' EVCA HHH );
*list smca;
CADWPVP='SOURCE' MODL1 (SMCA * CA * DWPM * USDT);
*
*---> (dwd/dT-dw/dT)*dT/dt
*
DWDMDWT='SOURCE' MODL1 ((DF1 - DWTM) * USDT);
*
*-----
* Apparent specific heat of concrete
*-----
*
*---> (roC)sec
* In caso di dipendenza dalla temperatura ins. qui
*---> W*Cw
WCW = CW * WW;
*list WCW ;
*---> Cad*dwd/dT
CADDWD= CAD*DWD;
*list CADDWD;
*---> -Ca*dw/dT
CADWT =-1. * SMCA * CA * DWT;
*---> (wd0-wd)*Cbw
WD0DCBW=(WD0-(TAV1.'WDM')) * CBW;
*list CADWT ;
*---> Apparent specific heat of concrete
ROCAP1=ROCSEC+CADDWD+WCW+CADWT+WD0DCBW;
ROCAP = 'CHAN' 'CHAM' ROCAP1 MODL1 ;
*list ROCAP

```

```

*
  KBET_SEC=K0_SEC  + (K1_SEC * (TAV1.'TMAX'));
  KBET  =NOMC 'K' (KBET_SEC * (1. + ((4./DEN_SEC)* WWW)));
  KBET=CHAN 'CHAM' KBET MODL1 'RIGIDITE';
  MABET= 'MATE' MODL1 'K' KBET 'RHO' 1. 'C' ROCAP  ;
*
*=====
*                THERMAL SOLUTION
*=====
*
'SI'(NON IFITER );
*
* Thermal capacity matrix
*
TCAPM='CAPACITE' MODL1 MABET;
A2= USDT * TCAPM ;
*
* Thermal stiffness matrix
*
RIG_M='CONDUCTIVITE' MODL1 MABET;
*
'SI' IFTC;
  KT = RIG_M 'ET' TCONDC;
'SINON';
  KT = RIG_M;
'FINSI';
  A1 =          LAMBDA * KT;
  A3 =(-1. * UMLAMBDA)* KT;
*
'SI' IFTB;
  MAT_RIGI = A1 'ET' A2 'ET' TCONDB ;
'SINON';
  MAT_RIGI = A1 'ET' A2 ;
'FINSI';
*
  B2 = A2 'ET' A3;
*
* Loads
*
  B1 = AGPGT + CADWPVP;
*
'SI' IFP ;
  B1 = CAJN_T  + B1;
'FINSI';
'SI' IFT ;
  B1 = B1 + TCONV ;
'FINSI';

```

```

  B3 = B2 * ('NOMC' 'T' T0S);
  MAT_CHPO = B1 + B3;
*
* Resolution
*-----
  T1 = 'RESOU' MAT_RIGI MAT_CHPO;
*-----
  TPS=T1S;
  T1S='EXCO' 'T' T1 'SCAL';
  VT0=(T1S-T0S)*USDT;
'FINSI' ;
*
*Valori fine iterazione
*
'SI' (FLG '>EG' 2);
'MESS' 'T1-MAX' ('MAXI' T1S) 'T1-MIN' ('MINI' T1S)
      'VT-MAX' ('MAXI' VT0) 'VT-MIN' ('MINI' VT0) ;
'FINSI';
*
*=====
*           MOISTURE TRANSFER SOLUTION
*=====
*
* Stiffness matrix
*
  CHA1 = 'CHAN' 'CHAM' AGDG MODL1 ;
  CHA2 = 'CHAN' 'CHAM' DWP MODL1 ;
*
  PMATR1 = 'MATR' MODL1 'RHO' 1. 'K' CHA1 'C' CHA2 ;
  PCOND1 = 'CONDUCTIVITE' MODL1 PMATR1 ;
*
* Capacity matrix
*
  PCAPA1 = 'CAPACITE' MODL1 PMATR1 ;
*
'SI' IFPC;
  KP = PCOND1 'ET' PCONDC;
'SINON';
  KP = PCOND1;
'FINSI';
  PA1 = KP * LAMBDA ;
  PA2 = PCAPA1 * USDT ;
  PA3 = -1. * UMLAMBDA * KP ;
  PB2 = PA2 'ET' PA3;
'SI' IFPB;
  PMAT_RIG = PA1 'ET' PA2 'ET' PCONDB;
'SINON';

```

```

    PMAT_RIG = PA1 'ET' PA2 ;
'FINSI';
*
* Loads
*
    PB1 = DWDMDWT ;
'SI' IFP;
    PB1 = PCONV + PB1;
'FINSI';
    PB3 = PB2 * ('NOMC' 'T' P0S);
*
    PMAT_CHP = PB1 + PB3 ;
*
* Resolution
*-----
    P1 = 'RESOU' PMAT_RIG PMAT_CHP ;
*-----
    PPS=P1S;
    P1S='EXCO' 'T' P1 'SCAL';
    VP0=(P1S-P0S)*USDT;
*
    WWW HHH = HTC_WWW T1S P1S TAV1 ;
    WD DF1 = HTC_CHBW T1S HHH TAV1 ;
    CHW = 'CHAN' 'CHAM' MODL1 WWW;
    MW = 'INTG' MODL1 CHW;
    CHD = 'CHAN' 'CHAM' MODL1 WD;
    MD = 'INTG' MODL1 CHD;
'SI' FDET;
'DETR' DF1;'DETR' CHW;'DETR' CHD;
'FINSI';
*
* -----
'SI' (FLG '>EG' 2);
'MESS' 'P1-MAX' ('MAXI' P1S) 'P1-MIN' ('MINI' P1S)
        'VP-MAX' ('MAXI' VP0) 'VP-MIN' ('MINI' VP0) ;
'FINSI';
* -----
'SI' (FLG '>EG' 3);
'MESS' 'H -MAX' ('MAXI' HHH) 'H -MIN' ('MINI' HHH)
        'W -MAX' ('MAXI' WWW) 'W -MIN' ('MINI' WWW) ;
'FINSI';
* -----
    ERPM1 ='MAXI'((P1S - PPS)/PPS) 'ABS' ;
    ERTM1 ='MAXI'((T1S - TPS)/TPS) 'ABS' ;
* -----
    GP_1 = 'GRAD' MODL1 ('NOMC' T P1S);
    GP_2 = 'CHAN' 'NOEUD' MODL1 GP_1;
  
```

```

      GP_3 ='CHAN' 'CHPO'  MODL1 GP_2;
* ----> J=-a grad p
      JFL='NOMC' LICF LICJ (-1.*AGDG * GP_3);
      'SI' FDET ;
      'DETR' GP_1; 'DETR' GP_2;'DETR' GP_3;
      'FINSI';
*
'SI' GRAF;
  titr ' t1' Istant1 ITER;
  trac maglia t1;
  titr ' p1' Istant1 ITER;
  trac maglia p1;
  titr 'hhh' Istant1 ITER;
  trac maglia hhh;
  titr ' www ' Istant1 ITER;
  trac maglia www;
*  titr 'jfl_x' Istant1 ITER;
*  trac maglia (exco JX_R jfl);
*  titr ' wd ' Istant1 ITER;
*  trac maglia wd;
'FINSI';
* -----
*PERDITA DI MASSA
*-----
      PMAS_T = 0.;
'SI' IFP;
      PPPT = (UMLAMBDA*P0)+(LAMBDA*P1);
'SI' ('EXIST' TAV1 'CONVECTION_HYDRAULIQUE' );
*  PCVCH = 'TIRE' PCHARCH IstantM ;
      PMAS_T=PMAS_T + ('MAXI' (('RESU' ((PCONDC * PPPT )- PCNVCH )) * T_PAS)
                      'AVEC' ('MOTS' 'Q'));
'FINSI';
'SI' ('EXIST' TAV1 'BLOCAGES_HYDRAULIQUE' );
      PMAS_T=PMAS_T + ('MAXI' (('RESU' ( PCONDB * PPPT)) * T_PAS)
                      'AVEC' ('MOTS' 'Q'));
*  PCVPI = 'TIRE' PCHARPI Istant1 ;
*  PMAS_T=PMAS_T + ('MAXI' (('RESU' ((PCONDB * P1 ) - PCVPI)) * T_PAS)
*                      'AVEC' ('MOTS' 'Q'));
'FINSI';
'SI' ('EXIST' TAV1 'FLUX_HYDRAULIQUE' );
*  PCVFH = 'TIRE' PCHARFH IstantM ;
      PMAS_T=PMAS_T - ('MAXI' (('RESU' PCNVFH) * T_PAS)
                      'AVEC' ('MOTS' 'Q'));
'FINSI';
'FINSI';
      PMAST1=PMAS_T+PERMAS;
      M0 = MT ;

```



```

MT = MW - MD + PMAST1 ;
ERM_ABS = ((MT - MI)/MI*100.);
ERM_REL = 'ABS'((MT - M0)/M0) ;
'SI' (FLG '>EG' 1);
'MESS' '-----'
      '-----';
'MESS' 'M.IN' MI '<= M.FW' MW '- M.RBW' MD '+ M.LW' PMAST1 '>=' MT ;
'FINSI';
'MESS' '-----'
      '-----';
'MESS' 'ER_M_P =' ERPM1 'ER_M_T =' ERTM1 'ERMAS_ABS_% =' ERM_ABS
      'ERMAS_REL =' ERM_REL ;
'MESS' '-----'
      '-----';
* -----
CCC = P1S 'MASQ' 'EGINFE' EPS1;
'SI' (('MAXI'CCC ) > .9);
'MESS' '==> LOW PRESSURE <==';
* P0 = (1. * CCC) + (P1EXT * (1. - CCC)) ; 'DETR' CCC;
'FINSI';
*
'SI' (ITER < 2) ;
  'ITER' OTO;
'FINSI' ;
'SI' IFITER ;
  'QUITTER' OTO ;
'FINSI' ;
*
'SI' ((ERPM1 <EG ERPM) 'ET' (ERTM1 <EG ERTM));
  IFITER = VRAI ;
'MESS' '==> CONVERGENCE <==';
*'QUITTER' OTO;
'FINSI';
*
'SI' (ITER 'EGA' N_ITER );
  IFITER = VRAI ;
'FINSI';
*
'FIN' OTO ;
*=====
*           FIN DES ITERATIONS
*=====
*           un peu de ménage
*
*           'MENAGE';
*
*=====

```

```

PERMAS=PERMAS + PMAS_T;
DWT0 = DWT ;
DWD0 = DWD ;
DWPO = DWP ;
DTMAX=('EXCO' 'T' T1 'SCAL')- TAV1.'TMAX';
MDTMAX=MASQUE DTMAX 'SUPERIEUR' 0. ;
TAV1.'TMAX' = TAV1.'TMAX' + ( DTMAX * MDTMAX);
TAV1.'WDM' = WD;
TAV1.'VT0' = VT0;
TAV1.'VPO' = VP0;
TAV1.'DWD0' = DWD0 ;
TAV1.'DWT0' = DWT0 ;
TAV1.'DWPO' = DWPO ;
*
*=====
*          SAUVEGARDE DES RESULTATS
*=====
N1_PAS=N1_PAS + 1;
'SI' (( EGA NN_PAS N1_PAS) OU ( EGA II N_PAS) OU (EGA II 1));
  IPAS=IPAS+1;
  'MESS' ' ';
  'MESS' 'SAUVEGARDE - PAS ' II ' INSTANT ' Istant1;
  TAV1.'RESULTATS'.IPAS = 'TABLE';
  TAV1.'RESULTATS'.IPAS.'INSTANT'           = Istant1;
  TAV1.'RESULTATS'.IPAS.'TEMPERATURE'       = T1 ;
  TAV1.'RESULTATS'.IPAS.'PRESSION'          = 'NOMC' LICT LICP P1;
  TAV1.'RESULTATS'.IPAS.'HUMIDITE_RELATIVE' = 'NOMC' 'HR' HHH;
  TAV1.'RESULTATS'.IPAS.'EAU_LIBRE'         = 'NOMC' 'WF' WWW;
  TAV1.'RESULTATS'.IPAS.'EAU_LIEE'         = 'NOMC' 'WD' WD ;
  TAV1.'RESULTATS'.IPAS.'MASSE_PERDUE'      = PERMAS;
  TAV1.'RESULTATS'.IPAS.'MASSE_FLUX'        = JFL ;
  TAV1.'RESULTATS'.IPAS.'MASSE_BILAN'       = MT ;
  TAV1.'RESULTATS'.IPAS.'MASSE_RELACHE'     = MD ;
  PSAT = 'NOMC' 'PSAT' (HTC_WTR TAV1 'SATU' T1);
  TAV1.'RESULTATS'.IPAS.'PRES_SAT'          = PSAT;
*
'SI' (EGA NN_PAS N1_PAS);
  N1_PAS=0;
  'FINSI';
'FINSI';
*
*
II = II + 1 ;
*
P0 = P1 ;
T0 = T1 ;

```

```

*
*=====
*
*           FIN DES PAS
*
*=====
*
'FIN' TOTO;
*
'FINPROC' TAV1;
$$$$ HTC_CHBW
* HTC_CHBW  PROCEDUR  ENEA      98/05/27    21:21:31    3201
*-----
*23456789012345678901234567890123456789012345678901234567890123456789012
*-----
*-----
* HTC_CHBW Dehydration curve
*
* Date: 16-03-98   mod : 21-07-11
*-----
'DEBP' HTC_CHBW  T0*'CHPOINT' H0*'CHPOINT' TAV1*'TABLE' ;
*-----
FLG1= TAV1.'FLG1';
'SI' FLG1; 'MESS' 'PROCEDURE-HTC_CHBW'; 'FINSI';
EWD =TAV1.'EWD';
WD1= 'VARI' T0 EWD 'SCAL';
WDM=TAV1.'WDM';
DF1= WD1 - WDM;
RESO1 = DF1 'MASQUE' 'SUPERIEUR' 0.;
RESO2 = H0 'MASQUE' 'EGINFE' 1.;
DF2=RESO1 * RESO2 * DF1;
* DF2=RESO1 * DF1;
'DETR' DF1;
'DETR' RESO1;
'DETR' RESO2;
WD =DF2 + WDM;
'FINPROC' WD DF2;
$$$$ HTC_PER
* HTC_PER  PROCEDUR  AM      09/12/15    21:15:17    6588
*-----
*23456789012345678901234567890123456789012345678901234567890123456789012
*-----
*-----
* HTC_PER Permeability of heated concrete
* (after BAZANT and THONGUTHAI)
* Date: 16-03-98 mod : 21-07-11
*-----

```

```

'DEBP' HTC_PER T0*CHPOINT HR*CHPOINT TAV1*TABLE ;
*-----
*
  FLG1= TAV1.'FLG1';
'SI' FLG1; 'MESS' 'PROCEDURE-HTC_PER'; 'FINSI';
  TREF=25.;
  GR=9.80665;
  HC=.75;
  A0=TAV1. 'A0';
  TRK = 273.15 + TREF;
  F2T95='EXP' (2700. * ((368.15 - TRK)/(TRK * 368.15)));
  AOP=A0*F2T95;
*
  RESO1= T0 'MASQUE' 'EGINFE' 95.;
  RESO2= T0 'MASQUE' 'SUPERIEUR' 95.;
  RESO3= HR 'MASQUE' 'EGSUPE' 1. ;
  RESO4= HR 'MASQUE' 'INFERIEUR' 1. ;
  RESO5= T0 'MASQUE' 'EGSUPE' TREF;
*
'SI' (('MINI' RESO1) > .9);
* T0 TUTTO UGUALE SOTTO 95
* ----> f2t
  TTT = 273.15 + T0 ;
  F2T = 'EXP' (RESO5 * 2700. * ((1. / TRK)-(TTT**-1)));
* CASO MISTO T SOTTO 95 E H SOTTO E SOPRA 1
*f1h dans le cas h >= 1
  F1HS = RESO3;
*f1h dans le cas h < 1
  ALFAT = (1. + ((19./(95.-TREF))*(95. - T0))**-1);
  F1HI = RESO4 * (ALFAT + ((1.-ALFAT)/(1.+(((1.- HR)/(1.-HC))**4))));
  F1H = F1HS + F1HI;
  AGDG=A0*F1H*F2T/GR;
*
'DETR' TTT ; 'DETR' ALFAT; 'DETR' F2T ;
'DETR' F1H ; 'DETR' F1HS ; 'DETR' F1HI ;
'DETR' RESO1; 'DETR' RESO2; 'DETR' RESO3;
'DETR' RESO4; 'DETR' RESO5;
'SINON';
'SI' (('MINI' RESO2) > .9);
* T0 TUTTO SOPRA 95
  T095= T0 - 95. ;
  F3T = 'EXP' (T095/(0.881 + (0.214*T095)));
  AGDG=AOP*F3T/GR;
'DETR' T095; 'DETR' F3T;
'DETR' RESO1; 'DETR' RESO2; 'DETR' RESO3;
'DETR' RESO4; 'DETR' RESO5;
'SINON';

```

```

*T0 SOPRA E SOTTO 95
  T095= RESO2 * ( T0 - 95.);
  F3T = 'EXP' (T095/(0.881 + (0.214*T095)));
*f1h dans le cas h >= 1
  F1HS = RESO3;
*f1h dans le cas h < 1
  ALFAT= (1. + ((19./ (95.-TREF))*(95. - T0)))** -1;
  F1HI =RESO4 * (ALFAT + ((1.-ALFAT)/(1.+(((1.- HR)/(1.-HC))**4))));
  F1H = F1HS + F1HI;
* ----> f2t
  TTT = 273.15 + T0 ;
  F2T = 'EXP' (RESO5 * 2700. * ((1. / TRK)-(TTT** -1)));
  AMA95=RESO2*(A0P*F3T);
  AIE95=RESO1*(A0*F1H*F2T);
  AGDG = (AMA95 + AIE95)/ GR;
'DETR' T095; 'DETR' F3T ;
'DETR' TTT ; 'DETR' ALFAT; 'DETR' F2T ;
'DETR' F1H ; 'DETR' F1HS ; 'DETR' F1HI ;
'DETR' AMA95; 'DETR' AIE95;
'DETR' RESO1; 'DETR' RESO2; 'DETR' RESO3;
'DETR' RESO4; 'DETR' RESO5;
'FINSI';
'FINSI';
*-----
'FINPROC' AGDG;
$$$$ HTC_WTR
* HTC_WTR  PROCEDUR  ENEA      98/05/27    21:21:39    3201
*-----
*23456789012345678901234567890123456789012345678901234567890123456789012
*-----
*-----
* HTC_WTR Properties of water
* Pressione di saturazione, densita,
* calore latente di evaporazione.
* Date: 16-03-98
*-----
'DEBP' HTC_WTR TAV1*'TABLE' M1*'MOT' TA*'CHPOINT' PA/'CHPOINT' ;
*-----
*
  FLG1= TAV1.'FLG1';
'SI' FLG1; 'MESS' 'PROCEDURE-HTC_WTR'; 'FINSI';
'SI' ( 'EGA' M1 'SATU');
* RESO1 = TA 'MASQUE' 'EGINFE' 374.15;
* RESO2 = 1. - RESO1 ;
*
* Saturation pressure in Mpa
*

```

```

'SI'('NON'('EXIS' TAV1 'EVPS'));
TPR1= 'PROG' 0. 4. 5. 6. 8. 10. 'PAS' 1. 36.
      38. 40. 'PAS' 5. 100. 'PAS' 10. 300.
      320. 340. 360. 374.14 2000.;
PS21='PROG' .000611 .000813 .000872 .000935 .001072 .001228 .001312
      .001402 .001497 .001598 .001705 .001818 .001938 .002064 .002198
      .002339 .002487 .002645 .002810 .002985 .003169 .003363 .003567;
PS22='PROG' .003782 .004008 .004246 .004496 .004759 .005034 .005324
      .005628 .005947 .006632 .007384 .009593 .01235 .01576 .01994
      .02503 .03119 .03858 .04739 .05783 .07014 .08455 .10133
      .1433 .1985 .2701 .3613 .4758 .6178 .7917 1.002
      1.254 1.554 1.906 2.318 2.795 3.344 3.973 4.688
      5.499 6.412 7.436 8.581 11.27 14.59 18.65 22.09 22.09;
PS2=PS21 'ET' PS22;
TAV1.'EVPS'= 'EVOL' 'MANU' TPR1 'SCAL' PS2 'SCAL' ;
'FINSI';
OUT= 'NOMC' 'SCAL' ('VARI' (TAV1.'EVPS') TA);
*
* AAA=33661.805;
* BBB=4744.058;
* OUT1= RESO1 *(AAA*('EXP'(((TA + 273.15)**-1)*(-1.*BBB))));
* OUT2= RESO2 *(AAA*('EXP'(((647.3)**-1)*(-1.*BBB))));
* OUT = OUT1 + OUT2;
*
'SINON';
'SI'('EGA' M1 'RHOW');
* Density as fun. of T and P
A0= 4.8863E-7 ;A1=-1.6528E-9 ;A2= 1.8621E-12;
A3= 2.4266E-13;A4=-1.5996E-15;A5= 3.3703E-18;
B0= 1.0213E3 ;B1=-7.7377E-1 ;B2= 8.7696E-3 ;
B3=-9.2118E-5 ;B4= 3.3534E-7 ;B5=-4.4034E-10;
PR= 2.E7;
B0=B0 - (PR*A0);B1=B1 - (PR*A1);B2=B2 - (PR*A2);
B3=B3 - (PR*A3);B4=B4 - (PR*A4);B5=B5 - (PR*A5);
RO1=TA*(TA*(TA*(TA*(TA*B5+B4)+B3)+B2)+B1)+B0;
OUT=PA*(TA*(TA*(TA*(TA*(TA*A5+A4)+A3)+A2)+A1)+A0)+RO1;
* mess 'density' (maxi out);
'SINON';
'SI'('EGA' M1 'HFG');
* Heat of vaporization J/Kg
*
RESO1 = TA 'MASQUE' 'EGINFE' 374.15;
*
OUT = RESO1 * 350000.*(('ABS'(374.15 - TA))**(1./3.));
'SINON';
'SI'('EGA' M1 'CW');
* Specific heat J/Kg

```

```

      OUT = 4100. ;
      'FINSI';
      'FINSI';
      'FINSI';
      'FINSI';
*
      'FINPROC' OUT;
$$$$ HTC_WWW
* HTC_WWW  PROCEDUR  AM          09/12/15    21:15:17    6588
*-----
*23456789012345678901234567890123456789012345678901234567890123456789012
*-----
*-----
* HTC_WWW Specific water content of concrete
* Sorption isotherms (after BAZANT and THONGUTHAI)
* Date: 16-03-98
*-----
'DEBP' HTC_WWW T0*CHPOINT P01*CHPOINT TAV1*'TABLE';
*-----
*
      FLG1 = TAV1.'FLG1';
      'SI' FLG1; 'MESS' 'PROCEDURE-HTC_WWW'; 'FINSI';
      EPS1 = TAV1.'EPSILON';
*Temperature °C
      TREF=25.;
*Pressure Mpa
      PREF=0.10135;
* Water specific volume [mc/kg] at 25 C and atmospheric pressure
      SV0=0.0010029;
* Water density [kg/mc] at 25 C and atmospheric pressure
      RHOW0=1./SV0;
* Concrete properties
*Kg/mc
      W1=TAV1. 'w1';
      CC=TAV1. 'c';
* Porosity at 25 °C
      N0=W1*SV0;
* Coefficient of linear thermal dilatation[ /°C]
      ALFA=TAV1.'ALFA';
* Bulk modulus
      EBM =TAV1.'EBM';
      BM  ='VARI' T0 EBM 'SCAL';
*
      PSAT=HTC_WTR TAV1 'SATU' T0;
*
      CCC=P01 'MASQUE' 'EGINFE' EPS1;
      'SI' (('MAXI' CCC) > .9);

```

```

'MESS' 'LOW OR NEGATIVE PRESSURE';
* MAGLIA = 'EXTR' MODL1 'MAIL';
* titr 'P01' Istant ITER ;
* trac MAGLIA p01;
  P0=(EPS1*CCC)+(P01*(1.-CCC));
'SINON';
  P0=P01;
'FINSI';
'DETR' CCC;
*
HR=P0/PSAT;
RESO1 = HR 'MASQUE' 'EGINFE' 0.93 ;
RESO2 = HR 'MASQUE' 'EGSUPE' 1.06 ;
RESO31 = 1. - ( RESO1 + RESO2 );
RESO3 = RESO31 'MASQUE' 'EGSUPE' 0.1 ;
H94 = 'MAXI' RESO1;
H15 = 'MAXI' RESO2;
H9415 = 'MAXI' RESO3;
*
WN=0.;WS=0.;WT=0.;
*
* water content wn at HR .9) 'OU' (H9415 > .9));
  TP = (( T0 + 10.) / ( TREF + 10.))**2.;
  MT = 1.04 - ( TP / ( 22.34 + TP ));
'FINSI';
*
'SI'(H94 > .9);
  EWC = 'LOG' ((W1/CC) * HR );
  WN1 = CC*('EXP'(EWC/MT));
  WN = RESO1 * WN1;
'FINSI';
*
* water content ws at HR >eg 1.06
*
'SI'((H15 > .9) 'OU' (H9415 > .9));
  WD AAA=HTC_CHBW T0 HR TAV1 ;DETR AAA;
  RHOW=HTC_WTR TAV1 'RHOW' T0 P0;
* N = N0;
* N = N0 + A (T0 -TREF);
  N =(N0 + ( WD / RHOW0));
  ALFT1 = 3. * ALFA * (T0-TREF);
'FINSI';
'SI'( H15 > .9);
  PH = 1. + ( 0.12 * ( HR - 1.04 ));
  NA = N * PH ;
  POMR = P0-PREF;
  RESO4= POMR 'MASQUE' 'EGSUPE' 0.;

```



```

WS1 = (1. + (((RESO4 * NA * POMR / BM) +
              ALFT1))) * NA * RHOW;
WS = RESO2 * WS1;
'FINSI';
*-----
* Water content wt for 0.93 < HR > 1.06
* Transition range by bezier spline interpolation
*-----
*
* Water content for H = 0.93 by first correlation
*
'SI'(H9415 > .9);
EWC93 = 'LOG' (w1 * 0.93 / CC) ;
WN931 = CC*('EXP'(EWC93*(MT**-1.))) ;
WN93 = RESO3 * WN931;
*
* Water content for H = 1 by first correlation
*
EWC100= 'LOG' (w1 * 1.0 / CC) ;
WN1001 = CC*('EXP'(EWC100*(MT**-1.))) ;
WN100 = RESO3 * WN1001;
*
* Water content for H = 1 by second correlation
*
P1 = 1.0 * PSAT;
RHOW1= HTC_WTR TAV1 'RHOW' TO P1;
P1MR = P1-PREF ;
RESO5 = P1MR 'MASQUE' 'EGSUPE' 0.;
WS1001 = (1. + (((RESO5 * N * P1MR / BM) +
                  ALFT1)))*N*RHOW1;
WS100 = RESO3 * WS1001;
*
* Water content for H = 1.06 by second correlation
*
P16 = 1.06 * PSAT;
RHOW16= HTC_WTR TAV1 'RHOW' TO P16;
P16MR = P16-PREF ;
RESO5 = P16MR 'MASQUE' 'EGSUPE' 0.;
WS161 = (1. + (((RESO5 * N * P16MR / BM) +
                  ALFT1)))*N*RHOW16;
WS16 = RESO3 * WS161;
*-----
* EQUAZIONI DELL' INTERP. DI BEZIER CON 4 PUNTI
*-----
'SI' ('NON'('EXIST' TAV1 'HINT'));
*
HHP = 'PROG' ;

```

```

  UUP = 'PROG' 0. PAS 0.001 1.;
  'REPE' CIC1 ('DIME' UUP) ;
  U = 'EXTR' UUP &CIC1;
*
* EQUAZIONE GENERALE PER L'ASCISSA H :
* h(t)=(h4 - 3h3 + 3h2 - h1)t**3 + (3h3 - 6h2 + 3h1)t**2
* + (3h2 - 3h1)t + h1
*
  HU = (0.13*(U**3)) - (0.21*(U**2)) + (0.21*U) + 0.93 ;
  HHP = 'INSE' HHP HU &CIC1 ;
  'FIN' CIC1;
  HHP=('PROG' 0.) 'ET' HHP 'ET' ('PROG' 20.);
  UUP=('PROG' 0.) 'ET' UUP 'ET' ('PROG' 0.);
  TAV1 . 'HINT' = 'EVOL' 'MANU' HHP 'SCAL' UUP 'SCAL';
  'FINSI';
*
  U = 'VARI' (TAV1 .'HINT') HR 'SCAL';
  WT1 = U*(U*(U*(WS16-(3*WS100)+(3*WN100)-WN93)+
    ((3*WS100)-(6*WN100)+(3*WN93)))+
    ((3*WN100)-(3*WN93)))+WN93;
  WT = RES03 * WT1 ;
  'FINSI';
*
* water content WW
*
  WW =WN+WS+WT;
*
  'FINPROC' WW HR;
$$$$

```

## 7 ALLEGATO 2: FILE DI INPUT CASI DI CONVALIDA

### 7.1 CASO-W1 MODULO DI ACCUMULO.

File primo lancio

```

*test assialsimmetrico:
* -scambio convettivo interno con fluido a 300
* -scambio convettivo esterno con aria a 25
* -senza perdita massa sul lato esterno
opti dime 2 mode axis elem qua8;
MESS 'SEI SICURO?';
  OPTI DONN 5;
TITOLO='test-w1';
opti impr (CHAIN TITOLO '.txt');
r1=.005 ; r2= 0.05; h1=0.005;
p1 = r1 0. ; p2= r1 h1;
l1 = d 1 p1 p2;
s1 = ((l1 tran dini 0.0001 dfin 0.001 ((0.3*r2) 0))
      tran dini 0.001 dfin 0.001 ((0.6*r2) 0))
      tran dini 0.001 dfin 0.0002 ((0.1*r2) 0) ;
l2=cote 2 s1;
l3=cote 3 s1;
l4=cote 4 s1;
trac s1;
opti donn 5;
*
mod_tot=modl s1 thermique isotrope qua8;
*convezione termica lato interno
* bt [w/m2-C] (t-te) [N/m2]
te1=300.;
bt1=100.;
tmodl1=modl l1 convection;
tmatl1=matr tmodl1 'H' bt1;
tconl1=conductivite tmodl1 tmatl1;
tcnv11=convection tmodl1 tmatl1 'T' te1;
ev_t1 =evol manu (prog 0. 6000.) (prog 1. 1.);
c_t11 =char tcnv11 ev_t1;
*convezione termica lato esterno
te3=25.;
bt3= 5.;
tmodl3=modl l3 convection;
tmatl3=matr tmodl3 'H' bt3;
tconl3=conductivite tmodl3 tmatl3;
tcnv13=convection tmodl3 tmatl3 'T' te3;
ev_t3 =evol manu (prog 0. 6000.) (prog 1. 1.);
c_t13 =char tcnv13 ev_t3;

```

```

*
* bw      [s/m] (p-pe) [N/m2]
*(bw*1.e6) [s/m] (p-pe) [MPa]
*convezione di massa lato esterno
pe3=1700.*1.e-6;
bw3=(0.2e-6)*1.e6;
pmodl3=modl  l3 convection;
pmatl3=matr pmodl3 'H' bw3;
pconl3=conductivite pmodl3 pmatl3;
pcnvl3=convection pmodl3 pmatl3 'T' pe3;
ev_p3 =evol manu (prog 0. 6000.) (prog 1. 1.);
c_pl3 =char pcnvl3 ev_p3;
*
tav1=table;
tav1.'TITR'=TITOLO;
tav1.'CONDUCTIVITE_THERMIQUE'=tconl1 et tconl3;
tav1.'CONVECTION_THERMIQUE'=c_tl1 et c_tl3;
tav1.'FRONTIERES_PRESSION'=table;
tav1.'FRONTIERES_PRESSION'. 1= table;
tav1.'FRONTIERES_PRESSION'. 1 .'MAILLAGE'=l3 ;
tav1.'FRONTIERES_PRESSION'. 1 .'CODIRXR' = 1.;
tav1.'FRONTIERES_PRESSION'. 1 .'CODIRYZ' = 0.;
tav1.'CONDUCTIVITE_HYDRAULIQUE'=pconl3;
tav1.'CONVECTION_HYDRAULIQUE'=c_pl3;
tav1.'LAMBDA'=0.55;
tav1.'GAMMA'=.5;
tav1.'EPSILON'=1.e-8;
tav1.'ERPM'=1.e-5;
tav1.'ERTM'=1.e-5;
tav1.'w1'=100.;
tav1.'c'=300.;
tav1.'F_STE'=0.24;
tav1.'F_INV'=1.;
tav1.'A0'=5.e-13;
tav1.'DEN_SEC'=2200.;
tav1.'CCP_SEC'=880.;
tav1.'K0_SEC' = 1.67;
tav1.'K1_SEC' =-0.00125;
tav1.'TEMPERATURE_INITIALE'=25.;
*pressione in MPa
tav1.'PRESSION_INITIALE'=1700.e-6;
tav1.'DETR' = vrai;
tav1.'GRAF' = FAUX;
tav1.'FLG1' = FAUX;
tav1.'FLG' = 0;
*
n_pas = 1000;

```

```

n_sa1 = 1000;
n_iter= 35;
dt = .1;
* 0. - 100. sec
HTCTRAN n_pas n_sa1 n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '1.sauve');
sauve tav1;
*
n_pas = 500;
n_sa1 = 500;
dt = 0.2;
* 100. - 200. sec
*
HTCTRAN n_pas n_sa1 n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '2.sauve');
sauve tav1;
n_pas = 3200;
n_sa1 = 400;
dt = 0.25;
* 200. - 1000. sec.
HTCTRAN n_pas n_sa1 n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '3.sauve');
sauve tav1;
fin;

```

File ultimo restart

```

*test assialsimmetrico
* -scambio convettivo interno con fluido a 300
* -scambio convettivo esterno con aria a 25
* -senza perdita massa sul lato esterno
opti dime 2 mode axis elem qua8;
MESS 'SEI SICURO ?';
OPTI DONN 5;
opti impr 'test-w1-r3.txt';
opti rest 'test-w19.sauve';
rest;
mod_tot=mod1 s1 thermique isotrope qua8;
*
n_pas = 2000;
n_sa1 = 500;
n_iter= 35;
dt = .2;
* 3000. - 3400. sec

```

```

HTCTRAN n_pas n_sal n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '10.sauve');
sauve tav1;
*
n_pas = 1500;
n_sal = 500;
n_iter= 35;
dt = .2;
* 3400. - 3700. sec
HTCTRAN n_pas n_sal n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '11.sauve');
sauve tav1;
*
n_pas = 1500;
n_sal = 500;
n_iter= 35;
dt = .2;
* 3700. - 4000. sec
HTCTRAN n_pas n_sal n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '12.sauve');
sauve tav1;
fin;

```

#### File di post trattamento

```

titolo='test-w1';
opti trac 'PSC' ftra (chain titolo '12.eps');
opti rest (chain titolo '12.sauve');
rest;
list tav1.resultats. 1 ;
*opti donn 5;
*
np=dime tav1 . resultats; mess np;
mper=prog;
ptem=prog;
repe cl1 np;
t11=tav1 . resultats . (&cl1-1) . temperature;
w11=tav1 . resultats . (&cl1-1) . eau_libre;
p11=tav1 . resultats . (&cl1-1) . pression;
h11=tav1 . resultats . (&cl1-1) . humidite_relative;
l11=tav1 . resultats . (&cl1-1) . eau_liee;
s11=tav1 . resultats . (&cl1-1) . pres_sat;
ptem=ptem et (tav1 . resultats . (&cl1-1) . instant);
mper=mper et (tav1 . resultats . (&cl1-1) . masse_perdue);

```

```

titr 'passo' (&c11-1) (tav1 . resultats . (&c11-1) . instant) ;
si (&c11 ega 1);
ev_t=evol chpo ttt 12;
ev_p=evol chpo ppp 12;
ev_w=evol chpo www 12;
ev_h=evol chpo hhh 12;
ev_l=evol chpo lll 12;
ev_s=evol chpo sss 12;
sinon;
ev_t=ev_t et (evol chpo ttt 12);
*dess ev_t;
ev_p=ev_p et (evol chpo ppp 12);
*dess ev_p;
ev_w=ev_w et (evol chpo www 12);
*dess ev_w;
ev_h=ev_h et (evol chpo hhh 12);
*dess ev_h;
ev_l=ev_l et (evol chpo lll 12);
*dess ev_l;
ev_s=ev_s et (evol chpo sss 12);
*dess ev_s;
finsi;
fin c11;
mess 'finale';
*dess (evol manu ptem mper);
dess ev_t titr 'temperatura';
dess ev_p titr 'pressione';
dess ev_w titr 'acqua libera';
dess ev_h titr 'umidità relativa';
dess ev_l titr 'acqua legata rilasciata';
dess ev_s;
*
```

## 7.2 CASO-K2 PROVINO CUBICO TRIDIMENSIONALE

File primo lancio

```

*caso tridimensionale
*scambio convettivo superficie esterna;
*
MESS 'SEI SICURO ?';
OPTI DONN 5;
opti dime 3 elem qua8;
TITOLO='test-K2';
opti impr (CHAIN TITOLO '.txt');
*dimensione cubo in m
l_c = 0.1;
x1 = l_c/2.;
p1 =0. 0. 0.;p1a =p1 plus ( -0.00105 -0.001 0.);
```

```

p2= x1 0. 0.;p2a =p2 plus ( 0.001 -0.001 0.);
p3= x1 x1 0.;p3a =p3 plus ( 0.001 0.00105 0.);
p4= 0. x1 0.;
pv= 0. 0. (100.*x1);
d1= d p1 p2 dini .02 dfin .0002 ;
d3 = plus d1 p4 ;
s1= regl d1 d3 dini .02 dfin .0002;
p3= point s1 proc p3;
d2 = cote 2 s1 ;
*trac pv s1;
ccci= d 1 p1 p2 d 1 p3 d 1 p1 ;
cccia= d 1 p1a p2a d 1 p3a d 1 p1a;
*opti donn 5;
nd = point s1 droi p1 p3 .00001;
*trac nd;
eld= elem s1 appu larg nd;
*trac eld;
celd = cont eld;
l13 = celd elem comp p1 p3;
*trac l13;
l31 = celd elem comp p3 p1;
*trac l31;
elds= eld elem appu larg l31;
*trac elds;
eldi= elem eld appu larg l13;
*trac eldi;
dif1= diff elds eldi;
*trac dif1;
dif2= diff eld dif1;
*trac dif2;
mess (nbel dif2);
*opti donn 5;
*
repe lop1 (nbel dif2);
eli= dif2 elem &lop1;
elt= chan eli tri6;
elti= elem 1 elt;
si (&lop1 ega 1); eltot= elti; sinon; eltot= eltot et elti; fin;
fin lop1;
*trac eltot;
*
s1tri= chan s1 tri6;
trac pv s1tri;
*
s2i=s1tri incl ccchia stri;
*trac pv ((cou1 ccchia vert) et s1tri);
*trac pv ((cou1 ccchia vert) et s2i );

```



```

*
i=0;
repe lop2 ((nbel eltot)/2);
i=i + 2;
eli= eltot elem i;
si ( i ega 2); eltot1= eli; sinon; eltot1= eltot1 et eli; finisi;
*trac (eltot1 et s2i);
fin lop2;
*trac eltot1;
*
sur1 = s2i et eltot1;
elim sur1 0.00001;
nb= nbno sur1;
ne= nbel sur1;
titr 'n.elm.' ne ' n.nod.' nb;
trac sur1;
opti elem pr15;
vol1=sur1 volu tran (0. 0. x1) dini 0.02 dfin 0.0002;
sur2=sur1 plus (0. 0. x1);
sur3=d2      tran (0. 0. x1) dini 0.02 dfin 0.0002;
elim (vol1 et sur2 et sur3) 0.00001;
sur4=sur2 et sur3;
trac (100. 200. 300.) vol1 face;
trac (100. 200. 300.) (sur1 et sur2 et sur3) ;
*
*opti donn 5;
*
mod_tot=modl vol1 thermique isotrope pr15;
*convezione termica lato esterno
* bt      [w/m2-C](t-te)[N/m2]
te1=200.;
bt1=50.;
tmods4=modl sur4 convection;
tmats4=matr tmods4 'H' bt1;
tcons4=conductivite tmods4 tmats4;
tcnvs4=convection tmods4 tmats4 'T' te1;
ev_t1 =evol manu (prog 0. 6000.) (prog 1. 1.);
c_ts4 =char tcnvs4 ev_t1;
* bw      [s/m](p-pe)[N/m2]
*(bw*1.e6)[s/m](p-pe)[MPa]
*convezione di massa lato esterno s2
pe3=1500.*1.e-6;
bw3=(0.2e-6)*1.e6;
pmods4=modl sur4 convection;
pmats4=matr pmods4 'H' bw3;
pcons4=conductivite pmods4 pmats4;
pcnvs4=convection pmods4 pmats4 'T' pe3;


```

```

ev_p3 =evol manu (prog 0. 6000.) (prog 1. 1.);
c_ps4 =char pcnvs4 ev_p3;
*opti donn 5;
*
tav1=table;
tav1.'MODE'=mod_tot;
tav1.'TITR'=TITOLO;
tav1.'CONDUCTIVITE_THERMIQUE'=tcons4;
tav1.'CONVECTION_THERMIQUE' =c_ts4;
tav1.'FRONTIERES_PRESSION' =table;
tav1.'FRONTIERES_PRESSION'. 1= table;
tav1.'FRONTIERES_PRESSION'. 1 .'MAILLAGE'=sur2 ;
tav1.'FRONTIERES_PRESSION'. 1 .'CODIRXR' = 1.;
tav1.'FRONTIERES_PRESSION'. 1 .'CODIRYZ' = 0.;
tav1.'FRONTIERES_PRESSION'. 1 .'CODIRZZ' = 0.;
*
tav1.'FRONTIERES_PRESSION'. 2= table;
tav1.'FRONTIERES_PRESSION'. 2 .'MAILLAGE'=sur3 ;
tav1.'FRONTIERES_PRESSION'. 2 .'CODIRXR' = 0.;
tav1.'FRONTIERES_PRESSION'. 2 .'CODIRYZ' = 0.;
tav1.'FRONTIERES_PRESSION'. 2 .'CODIRZZ' = 1.;
*
tav1.'CONDUCTIVITE_HYDRAULIQUE'=pcons4;
tav1.'CONVECTION_HYDRAULIQUE'=c_ps4;
tav1.'LAMBDA'=0.55;
tav1.'GAMMA'=.5;
tav1.'EPSILON'=1.e-8;
tav1.'ERPМ'=1.e-4;
tav1.'ERTM'=1.e-4;
tav1.'w1'=100.;
tav1.'c'=300.;
tav1.'F_STE'=0.24;
tav1.'F_INV'=1.;
tav1.'A0'=5.e-13;
tav1.'DEN_SEC'=2200.;
tav1.'CCP_SEC'=880.;
tav1.'K0_SEC' = 1.67;
tav1.'K1_SEC' =-0.00125;
tav1.'TEMPERATURE_INITIALE'=25.;
*pression in MPa
tav1.'PRESSION_INITIALE'=1500.e-6;
tav1.'DETR' = vrai;
tav1.'GRAF' = FAUX;
tav1.'FLG1' = FAUX;
tav1.'FLG' = 0;
*
n_pas = 500;


```

```
n_sa1 = 500;
n_iter= 35;
dt = .02;
* 0. - 10. sec
HTCTRAN n_pas n_sa1 n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '1.sauve');
sauve tav1;
*
n_pas = 500;
n_sa1 = 500;
n_iter= 35;
dt = .02;
* 0. - 20. sec
HTCTRAN n_pas n_sa1 n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '2.sauve');
sauve tav1;
*
n_pas = 1000;
n_sa1 = 500;
n_iter= 35;
dt = .02;
* 0. - 40. sec
HTCTRAN n_pas n_sa1 n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '3.sauve');
sauve tav1;
*
n_pas = 3000;
n_sa1 = 500;
n_iter= 35;
dt = .02;
* 0. - 100. sec
HTCTRAN n_pas n_sa1 n_iter dt mod_tot tav1 ;
*
opti sauve ( chain (tav1.'TITR') '4.sauve');
sauve tav1;
*
fin;
```

 <b>Ricerca Sistema Elettrico</b>	<b>Sigla di identificazione</b>	<b>Rev.</b>	<b>Distrib.</b>	<b>Pag.</b>	<b>di</b>
	NNFISS – LP1 - 019	0	L	58	59

## 8 BIBLIOGRAFIA

- [1] G.M.Giannuzzi, Heat and Mass Transfer in Concrete at High Temperature: Theoretical Manual of HTCTRAN procedure, ENEA CT-SBE-00014;1998
- [2] G.M.Giannuzzi, Validation of HTCTRAN procedure, ENEA CT-SBE-00012;1998
- [3] Z.P. Bazant and M.F. Kaplan, Concrete at High Temperatures : Material Properties and Mathematical Models, Longman (1996)
- [4] Z.-X. Gong and A. S. Mujumdar, Development of Drying Schedules for One-Side-Heating Drying of Refractory Concrete Slabs Based on a Finite Element Model, J.Am.Ceram.Soc.,79[6] 1649-58(1996).
- [5] P. Majumdar,A. Gupta, A. Marchertas, Moisture propagation and resulting stress in heated concrete walls, Nuclear Engineering and design 156 (1995),147-158
- [6] M. V.G. de Morais, P. Pliya, A. Noumowe, A.-L. Beaucour, S. Ortola, Contribution to the explanation of the spalling of small specimen without any mechanical restraint exposed to high temperature, Nuclear Engineering and Design 1240 (2010),2655-2663
- [7] F.Corsi, G.M.Giannuzzi, G.Khoury, C.Majorana, B.Schrefler, P.Kalifa, M.Cheyrezy, M. Behloul, B.Aarup, Scientific Summary of the HITECO Research BRITE Programme,ACI Int Conf on “State-of-the-art in high performance concrete” Chicago;1999
- [8] F.Pesavento, G.M. Giannuzzi- HITECOSP2 - Theoretical Manual - ENEA-UNIPD; 2006
- [9] F.Pesavento, G.M. Giannuzzi- HITECOSP2 - User Guide - ENEA-UNIPD; 2006
- [10] R.W. Lewis, B.A. Schrefler, The finite element method in the static and dynamic deformation and consolidation of porous media, Wiley (1998) 181-184
- [11] M. Bomberg, C.J. Shirliffe, Influence of moisture and moisture gradients on heat transfer through porous building materials,ASTM STP 660, 1978
- [12] Nicos S. Martys, Survey of concrete transport properties and their measurement, NISTIR 5592, 1995
- [13] Daniel R. Flynn, Response of high performance concrete to fire conditions : Review of thermal property data and measurement techniques, NIST GCR 99-767
- [14] J.C. Maréchal, Contribution à l'Etude des Propriétés Thermiques et mécaniques du Béton en fonction de la Température, Annales de l'Institut Technique du Batiment et des Travaux Publics No. 274 (Oct. 1970), p. 121.

 <b>Ricerca Sistema Elettrico</b>	<b>Sigla di identificazione</b> NNFISS – LP1 - 019	<b>Rev.</b> 0	<b>Distrib.</b> L	<b>Pag.</b> 59	<b>di</b> 59
--	---	------------------	----------------------	-------------------	-----------------

- [15] R.w. Lewis et altri, The Finite Element Method in Heat Transfer Analysis. Wiley, 1996.